
1. Neprocedurální programování 
Převážná většina programovacích  jazyků dnes běžně užívaných má,  přes  velké  rozdíly,  jeden  podstatný  rys  -  

jde  o tzv. p r o c e d u r á l n í  j a z y k y . Popisujeme v  nich (víceméně přesně), j a k  s e  m á   d a n á   
ú l o h a   v y ř e š i t .   Základem,  na  kterých tyto jazyky stojí, je pojem p r o m ě n n é  jako  o z n a č e n í  
m í s t a  v   p a m ě t i  a   p ř i ř a z o v a c í h o   p ř í k a z u ,  který umožní  nějak získanou hodnotu  uložit do 
vhodné  proměnné. V tom, co  je nad těmito  základy vystavěno, se  Basic, assembler, Fortran, Algol,  Pascal, Ada, PL/I, 
C  a C++ (abychom vyjmenovali jen  ty  nejznámější)  liší  velmi  podstatně,  daný  rámec  však  nepřekračují. 

Existují  však  i  programovací  jazyky  založené  na jiných principech.   V  n e p r o c e d u r á l n í m    
p r o g r a m o v a c í m   j a z y c e   (trochu zjednodušeně  řečeno)  s p e c i f i k u j e m e   p r o b l é m , 
který  chceme řešit  a n e  n u t n ě  p ř e s n ý  z p ů s o b ,  j a k ý m  s e  m á  ř e š i t . Historicky  
nacházejí takové   jazyky  uplatnění   především  v   oblasti  tzv.   umělé inteligence,  která je  mimo jiné  
charakterizována převahou úloh nenumerického  charakteru. Nejdůležitějšími  představiteli těchto jazyků jsou LISP a 
Prolog. 

LISP je  velmi starý programovací jazyk,  který vznikl v USA již v  padesátých letech. Slouží ke  zpracování 
seznamů, na které převádí všechny  datové struktury. LISP  je dodnes populární,  má mnoho dialektů (pravděpodobně 
ještě více  než Basic), můžete se s ním  setkat např.  v kreslicím  systému Autocad,  většina úplných implementací  
jazyka  LOGO  obsahuje  i "lispovskou"  část.  Pěknou a dobře čitelnou knihou o LISPu je kniha [1]. 

My jsme  si vybrali druhý  z této dvojice,  jazyk P r o l o g . Je představitelem tzv. l o g i c k é h o  
p r o g r a m o v á n í   (i když, jak uvidíme, tento rámec poněkud přesahuje). Jak uvidíme, neskládá se p r o g r a m  
v  P r o l o g u  z příkazů, ale j e  v l a s t n ě  p o p i s e m  r e l a c í  p o m o c í  j e d n o d u c h ý c h  
f o r m u l í .  Na  v ý p o č e t  programu v Prologu se pak lze  dívat  jako  na  h l e d á n í   d ů k a z u   vhodné 
formule tzv. Hornovy  logiky  rezoluční  metodou.  Nemusíte  však  mít  strach z přílišného formalismu, P r o l o g  j d e  
v y k l á d a t  i  b e z  p ř í m ý c h  o d k a z ů  n a  l o g i k u .  Takový způsob výkladu jsme zvolili i my. 

Ovládnutí  myšlenkových základů  programování v  Prologu Vám pomůže  lépe   pochopit  i  programování   
"klasické",  především rekursi, která hraje v Prologu - stejně jako v LISPu zcela zásadní roli. Uvidíte také, že přes  
všechny rozdíly  mezi procedurálními a neprocedurálními programovacími jazyky, je programování přece jen "jen  
jedno" - většinu skutečně chytrých myšlenek z jednoho světa můžeme použít i ve světě druhém. Ze zkušenosti se 
studenty MFF vím, že takový zcela odlišný pohled na programování je velmi užitečný pro formování názorů na 
programování. 

 C í l e m  našeho výkladu  n e n í  tedy pouze P r o l o g  s á m , ale především pomoci Vám 
p o c h o p i t ,  c o  t o  p r o g r a m o v á n í  v l a s t n ě  j e .  

Náš postup výkladu je blízký knize [2], která je velmi dobře čitelná,  ale  hůře  dostupná  -  neexistuje  český ani 
slovenský překlad.  Nejdostupnější  -  a  velmi  dobrou  - je učebnice [3], používá  však jinou  metodu výkladu,  blíže 
k logice, a obsahuje menší  množství příkladů. 

2. Jednoduchý příklad 

Bez  dalších  úvodů  si  ukážeme  rovnou  jednoduchý program v Prologu. Bude sloužit k  zodpovídání otázek 
o rodinných vztazích malé  komunity lidí.  Představme si,  že Spolek  klepen (dále jen SK) vyslal jednu  ze svých 
zasloužilých  členek jako vyzvědačku,  aby zjistila  "kdo co  s kým".  Zprávu vyzvědačky  si můžete  přečíst v pravém  
sloupci,  který  je  oddělen  znaky  % . 

 V levém sloupci vidíme program v Prologu,  který vystihuje stejnou skutečnost (ve skutečnosti  je  celý  
následující  text  regulérním prologovským programem, protože celý text  za znakem %  až do konce  řádky se v Prologu 
chápe  jako komentář.  Čtěte zatím  jen vyzvědaččinu  zprávu vpravo a na vlastní program koukejte jen po očku. 

muz(adam).  %   Tam vám byl jeden muž Adam 
manz(adam,eva). %  ti dva byli manželé. 
rodic(adam,josef). %  Adam byl rodičem Josefa 
rodic(adam,hugo). %  a Huga. 
rodic(eva,josef). %  Rodičem Josefa byla i Eva, 
rodic(eva,hugo). %  zrovna tak byla i rodičem Huga. 

muz(karel).  %  Byl tam i další muž, jmenoval se Karel, 
zena(helena). %  a žena Helena. 
manz(karel,helena). %  Ti také byli manželé. 
rodic(karel,zuzana). %  Karel byl rodičem Zuzany, 
rodic(helena,zuzana). %  i Helena byla jejím rodičem. 
rodic(helena,alfred). %  Helena byla i rodičem Alfreda 
rodic(karel,alfred). %  a Karel taky. 

zena(klara).  %  Byla tam i žena Klára, 
rodic(klara,emil). %  která byla rodičem Emila, 
muz(emil).  %  což byl muž. 
rodic(karel,emil). %  Rodičem Emila byl i Karel 

zena(kunhuta). %  Byla tam i žena Kuhnuta, 
manz(zibrid,kunhuta). %  jejíž manželem byl Žibrid. 
muz(zibrid).  %  Byl to muž. 
rodic(kunhuta,eva). %  Kunhuta byla rodičem Evy 
rodic(kunhuta,katka). %  a Katky. 
zena(katka).  %  Katka byla žena. 
rodic(zibrid,eva). %  Rodičem Evy 
rodic(zibrid,katka). %  a Katky byl také Žibrid. 

zena(ruzena).          %  Žena Růžena 
rodic(ruzena,jan).     %  byla rodičem Jana. 
manz(jan,lucie).       %  Jan byl manželem Lucie. 
muz(jan).              %  Jan byl muž 
zena(lucie).           %  a Lucie žena. 
rodic(jan,adam).       %  Oba byli 
rodic(lucie,adam).     %  rodiči Adama. 

muz(jose).             %  Muž José 
manz(jose,katka).      %  byl manželem Katky. 

Zpráva   klepny-vyzvědačky  je   psána  poměrně   neobratně a rozhodně  ne  krásným  jazykem,  přesto  však  z ní  
lze kromě trivialit jako, že 

�� Jan a Lucie jsou manželé 
�� Helena byla žena 

vyčíst i lecos zajímavého. Například, že 
�� manžel  Heleny Karel  měl nemanželské  dítě Emila  s nějakou Klárou, 

�� Hugova teta Katka byla manželkou Josého, neví se však o tom, že by některý z nich měl nějaké dítě 
�� Adamova tchýně se jmenovala Kunhuta. 

Jistě vám nečiní potíže se o tom sami přesvědčit. Zkuste to ! 

Cvičení: 

 Zjistěte 
�� jak se jmenoval manžel Josefovy babičky 
�� kolik vnuků měl Žibřid. 

To bylo snadné. Pravděpodobně by vás ale přivedla do rozpaků zakázka SK,  abyste jim udělali  program, který by  
takové závěry umožnil činit. Jak máme naprogramovat, co to je tchýně ? 

Budeme-li předpokládat, že SK chce k zodpovídání dotazů použít počítače,  musíme se na  to, co klepna-
vyzvědačka vlastně zjistila, podívat blíže  -  a trochu to zformalizovat . 

Klepna  zjistila,  že  některé  objekty  zkoumání  jsou muži a jiné ženy.  Dále zjistila, že  mezi některými objekty  
je vztah manželství a že některé objekty jsou rodiči jiných. 

Použijeme-li jazyka matematické  logiky, můžeme říci, že k popsání situace použila vyzvědačka čtyř predikátů: 

     predikát                 význam 

     muz(X)                 X je muž 
     zena(X)                X je žena 
     manz(X,Y)              X a Y jsou manželé, (X manžel, Y manželka) 
     rodic(X,Y)             X je rodičem Y 

první dva  predikáty mají jeden argument (jsou  unární) a reprezentují tedy vlastnosti objektů, druhé dva mají dva 
argumenty (jsou binární) a vystihují tedy vztah dvou objektů. 

Podíváme-li se na levý sloupec - program v Prologu - vidíme, že  je  vlastně  výčtem   tvrzení  (formulovaných  
pomocí  těchto predikátů),  které se  klepně  o  dané komunitě  podařilo zjistit. Dalším  rozdílem proti vyprávění klepny 
v pravém sloupci je,  že  ze  syntaktických  důvodů  musí jména objektů  v prologovském programu začínat  malým  
písmenem a nesmí se v něm používat  specifické znaky české  abecedy, jako jsou č , ž a ě. 

Klepna-vyzvědačka   mohla   skutečnosti,   které   zjistila, vyjádřit  i  pomocí  jiných  predikátů. Mohla 
například místo predikátu rodic(X,Y) použít dvou predikátů matka(X,Y)  a otec(X,Y). Tím by si ušetřila 
nutnost konstatovat o některých objektech, že jsou to muži  resp. ženy.  Predikáty muz  a zena  by však při tom zcela 
odstranit nešlo - bez nich bychom nemohli popsat pohlaví  těch, kteří  nejsou rodiči. 

Jistě byste  si dokázali vymyslet  i jiné množiny  predikátů dostatečných k vyjádření dané skutečnosti. Zkuste to! 

Jak však s takovými  informacemi má pracovat počítač? Klepny (spolu s námi) dobře vědí, co  je to tchýně nebo 
teta. Domyslí se také, že Hugo bude asi muž,  i když to ze zprávy vyzvědačky přímo nevyplývá. Bude-li program 
obsahovat dvojici faktů 

rodic(a,b). rodic(b,a). 

poznají, že to není pikanterie, ale nesmysl. 

Jak  však takovým  vědomostem "naučit  počítač" ?  Řešení je překvapivě snadné - v ů b e c   h o  t o  u č i t  
n e b u d e m e .  

R e z i g n u j e m e    z c e l a    n a    v ý z n a m    p r e d i k á t ů ,  se  kterými pracujeme, a  
b u d e m e  s  n i m i   p r a c o v a t  j e n  j a k o   s  " n á p i s y " ,  k t e r é  m a j í   s v o j i   
s t r u k t u r u ,   n e   v š a k   v ý z n a m ,   k t e r ý  b y  m o h l  p r o g r a m  v y u ž í t .   Všechny  
vlastnosti,  které  o  predikátech  a objektech předpokládáme,  musíme  v prologovském  programu  specifikovat. 
Takovému  postupu  se  říká  f o r m á l n í   a  je jednou ze základních pracovních metod matematiky a logiky. 

Logika  se  zabývá  takovými  souvislostmi mezi  jednotlivými  výroky,  které  vyplývají  jen  z jejich  
struktury, ne z jejich významu. To jí umožňuje odvozovat "obecně platné" závěry. 

Výhody formálního postupu znáte však i z "běžného života".  Bylo velkým pokrokem,  když  (obsahové)  
hieroglyfické  písmo  bylo nahrazeno hláskovým.  Slovo  "slunce"  sice  cizinci  nemusí  (narozdíl  od hieroglyfu -  
obrázku sluníčka )  být srozumitelné, ale  zase jde vytvořit slovník, ve kterém lze najít význam slov, kterým 

nerozumíme. Zkuste hledat v  čínském slovníku - to je fuška a věda.  Formalizace  zápisu  řeči  podstatně zjednodušuje, 
ne-li dokonce umožňuje, knihtisk a zpracování informace. 

Jiným příkladem  úspěšné formalizace je  (desítková) poziční soustava,  která  podstatně  zjednodušuje  početní  
operace i pro člověka (o počítačích nemluvě). Ale zpátky k Prologu. 

Programy  v  Prologu  pracují  s  a t o m y   pojmenovanými slovy začínajícími  malými písmeny  (např. hugo,  
emil, manz),  aniž by "rozuměly", co pro nás tyto objekty představují. Z tvaru programu se  pozná jen,  že atomy  
hugo a  emil označují  objekty (nulární predikáty) a atom manz označuje binární predikát.  

Základní jednotkou prologovského  programu je tzv. k l a u z u l e , která  vždy  končí  tečkou.   Náš  program  je  
velmi  jednoduchý a obsahuje  jen  nejjednodušší  typ  klauzulí  tzv.  f a k t y .  Fakt v prologovském programu  je 
konstatováním, že  pro nějaké objekty platí nějaké predikát. 

Prolog je interaktivní jazyk,  je-li program (například náš) zaveden v paměti,  můžeme se ho ptát.  Na obrazovce 
se  objeví výzva  ? -   a  počítač čeká na náš dotaz. 

Ukažme si příklady takové konverzace. Naše dotazy  budeme psát kursivou, odpovědi počítače s odsazením a  
tučně. Jako dříve budeme v pravé části řádky psát za znakem "% " význam dotazů a odpovědí. Přijetí odpovědi 
počítače musíme potvrdit znakem „enter“ - odeslání řádku. Tak dostaneme znovu výzvu ?- a můžeme se ptát znovu. 

?-manz(jan,lucie). %  Jsou Jan a Lucie manžely? 
 yes.            %  Ano, jsou. 
?-manz(lucie,jan). %  Jsou Lucie a Jan manžely? 
 no.             %  Nejsou. 

V prvním případě byla odpověď  ano, protože v našem programu je uložena  jen informace manz(jan,lucie).  
O tom, že  by platilo manz(lucie,  jan), v  něm však  nic není.  A jak  jsme si  řekli, Prolog  nic  neví  o  tom,  že  my  
v některých případech vztahy manz(X,Y) a manz(Y,X) považujeme za ekvivalentní. 

Pomocí dotazů tohoto typu si  můžeme pouze ověřovat, zda pro přesně  specifikované  objekty  platí  dané  
predikáty. Kdybychom mohli  své  dotazy  formulovat  jen  takto,  mělo  by to jen malý užitek. SK by také  
pravděpodobně nebyl takovým programem nadšen. Ve  svých  dotazech  prologovskému  programu  můžeme  však 
použít i proměnných (nepředstavujte si však nic podobného proměnné např. v Pascalu) .  Proměnná se v  Prologu pozná 
tak,  že začíná velkým písmenem. 

?-manz(X,lucie).  %  Kdo je manželem Lucie ? 
 X = jan.            %  Manželem Lucie je Jan. 

Jak Prolog vyhodnocoval tento dotaz?  

Hledal takový objekt T, který by  mohl substituovat (dosadit)  za proměnnou X, aby platilo tvrzení manz(T,lucie). 
A našel jako první substituci  X = jan , která vyhovovala. 

Můžeme se ale také zeptat 

?-manz(jan,X). %  Kdo je manželkou Jana ? 
 X = lucie. 

Vidíme tedy, že samotný program v  Prologu n e m á  u r č e n  " s m ě r  v ý p o č t u "  tj.  co je  vstup a  
co má  být výstupem.  Tento směr  udává až dotaz, který uživatel zadá. 

P r o g r a m  j e  t e d y  j e n   s o u h r n e m  f a k t ů ,  k t e r ý  " n e l z e  s p u s t i t " .  V ý p o č e t  s t a r t u j e m e  a ž  
z a d á n í m  d o t a z u .  

Odpověď  na  následující  dotaz  bude  pochopitelně záporná, protože lucie  není prvním argumentem  žádného 
faktu o  predikátu manz. 

?- manz(lucie,X). %  Kdo je manželkou Lucie ? 
 no.      %  Nemá ji. 

Zkusme zadat tento dotaz 

?-rodic(X,katka). %  Kdo je rodičem Katky ? 
 X = kunhuta  %  Kunhuta 

V programu jsou obsaženy dva relevantní fakty 

rodic(kunhuta,katka). a   rodic(zibrid,katka). 

Obdrželi  jsme  odpověď  X = kunhuta  proto,  že  tento fakt se vyskytuje  v programu  dříve. Pokud  nás zajímají  
i další  možné substituce,  které  vedou  k  pravdivému  tvrzení,  máme  možnost požádat o další pokusy o jejich nalezení  
tak, že na odpověď nepotvrdíme znakem „enter“ jako jsme činili dosud, ale vyžádáme hledání dalších substitucí znakem 
středník.  Zopakujme tedy poslední konverzaci: 

?-rodic(X,katka). %  Kdo je rodičem Katky ? 
 X = kunhuta  %  Kunhuta - chci další možnosti, 
    ;            %  proto napíši středník 
 X = zibrid %  Prolog našel další možné řešení - Žibrid 
    ;    %  chci další možnosti, 
     no                      %  jiného rodiče katka již nemá 

Středník  způsobí  odvolání   posledně provedené substituce  a vyvolá pokus o hledání  další   vyhovující  
substituce.  Jistě   vám  to  něco připomíná -  b a c k t r a c k i n g . Skutečně, jak  později uvidíme, jádrem algoritmu 
interpretace prologovských  programů je postup "ú p l n é h o  p r o c h á z e n í  d o  h l o u b k y  s  n á v r a t y ", tedy 
backtrackingu. 

Můžeme  také  zjišťovat  všechny  manželské  dvojice  pomocí následující konverzace: 

?-manz(On,Ona).                  %  Která dvojice jsou manželé? 
 On = adam , Ona = eva  ;   %  Adam a Eva, a další ? 
 On = karel , Ona = helena  ;  %  Karel a Helena, a další ? 
 On = zibrid , Ona = kunhuta ; %  Žibřid a Kunhuta, a další ? 
 On = jan , Ona = lucie  ;     %  Jan a Lucie , a další ? 
 On = jose , Ona = katka  ;  %  José a Katka , a další ? 
 no                           %  žádná další manželství 
                                 %  nejsou v programu registrována 

Naše dotazy mohou být i složitější. Napíšeme-li v dotazu dva predikáty  za sebou,  odděleny čárkou,  hledáme 
substituci, která způsobí platnost obou. Například 

?-rodic(X,katka),muz(X). %  Kdo je rodičem Katky a mužem 
   %  zároveň, tj. jejím otcem ? 
X = zibrid ;  %  Otcem je Žibrid, 
     no           %  další již není k dispozici. 

U  tohoto  dotazu  je  třeba  si  opět vysvětlit, jak Prolog dospěl  ke své  odpovědi. Nejprve  se snažil  nalézt za 
proměnnou X takovou substituci, aby splnil predikát rodic(X,katka). Jak již víme, našel  nejprve substituci X = kunhuta.  
Tím byl splněn  první  cí l  -   platnost  predikátu rodic(X,Katka)  - a Prolog  mohl přejít k pokusu o  splnění cíle 
druhého -  muz(kunhuta) (za proměnnou X je již substituováno).  Ten se však  (pochopitelně) splnit nepodařilo. 

  Proto Prolog zahájil návrat  - odvolal  poslední substituci za  X (obdobně  jako činí po  středníku) - a  pokusil  se  
splnit  cíl  rodic(X,katka)  jinak. Našel další možnost substituce X = zibrid,  která již vedla i ke splnění druhého 
požadovaného cíle. Po dalším návratu vyvolaném zadáním středníku uživatelem, již nebyla jiná substituce splňující oba 
cíle nalezena. 

Viděli  jsme tedy,  že operátor  č á r k a  má  v Prologu  význam k o n j u n k c e  -  logické  spojky & (a) ,  
která vyjadřuje  současnou platnost  výroků, které  spojuje. Prolog  se snaží  této současné platnosti dosáhnout 
postupným splňováním jednotlivých cílů odleva (s případným návratem při neúspěchu). 

 Program,  se kterým  jsme dosud  pracovali, obsahoval  pouze fakty.  V  prologovských  programech  však  
mohou  být i klauzule složitější,  tzv.  p r a v i d l a .  Pravidlo  vlastně definuje platnost jednoho  predikátu  pro  
nějaké   objekty  na  základě  platnosti (jiných) predikátů pro (jiné) objekty. 

Například  do našeho  programu bychom  mohli přidat pravidlo definující nový predikát otec: 
 

otec(Otec,Dite):- rodic(Otec,Dite),muz(Otec). 
   %  otcem Dítěte je jeho rodič Otec, který je také muž. 

Čárka, stejně jako dříve v dotazech, znamená i na pravé straně pravidel konjunkci. 

Povšimněte si také toho, že „argumenty“ predikátů v předchozí klauzuli jsou proměnné, nikoli jména objektů. Je to 
proto, že klauzule nevyjadřuje žádné tvrzení o jednotlivých objektech, ale je vlastně definicí predikátu otec. 

Po přidání  této klauzule, definující predikát otec, do  programu, bychom mohli  nového predikátu použít v dotazu: 

?-otec(X,katka).   %  Kdo je otcem Katky? 
     X = zibrid ;   %  otcem je Žibrid, 
     no            %  další již není k dispozici. 

V rámci rovnoprávnosti přidejme do programu i klauzuli definující, co je to matka. 

matka(Matka,Dite):- rodic(Matka,Dite), zena(Matka). 
   %  Matka Dítěte je jeho rodič, který je žena. 

Nyní již  také můžeme definovat  predikát manzdite(X), který vyjadřuje, že X je manželským dítětem: 

manzdite(X):- otec(Ot,X),matka(Mat,X),manz(Ot,Mat). 
           %  Dítě je manželské, jestliže jeho otec a matka jsou  manželé 

Z  takového predikátu  by již  členky SK  mohly mít radost ! Asi  by jim  ji však  zkazila skutečnost,  že definovat  
predikát nemanzdite(X), který  by je zajímal nepochybně  víc, je v Prologu dost těžké a  nemáme pro to zatím dostatečně 
silné prostředky.  Zkuste se zamyslet nad tím, proč ! 

Pro pedanty  (či životem zkoušené)  připomínáme, že jsme  do našeho modelu nezahrnuli rozvody.  Vadí-li vám to, 
zkuste program příslušně modifikovat ! 

Zastavme  se  na  chvíli  u  syntaxe  Prologu.  Jak už víme, začínají jména  atomů malým písmenem  a 
proměnné velkým.  Je tedy "matka" jméno predikátu a "Matka" jméno proměnné. 

Zkusme  na  závěr  této  kapitoly  definovat několik dalších nových predikátů: 

deda(D,V):- rodic(R,V),otec(D,R). 
   %  je-li D otec rodiče R, pak je dědou V 
teta(T,X):- rodic(Y,X), sestra(T,Y). 
   %   Je-li Y rodič X a T je jeho sestra, je T tetou X 
sestra(Ses,X):- rodic(Y,Ses),rodic(Y,X),zena(Ses). 
   %  Má-li Ses  stejného rodiče jako X a je-li to žena, 
   %  pak je to sestra X 

Jsou tyto predikáty naprogramovány správně ? 

N e  !   Podle  této definice  je každá  žena (pokud  se ví o alespoň  jednom z jejích rodičů) svojí sestrou. 

Abychom mohli sestru definovat správně, potřebujeme predikát 

ruzne(X,Y)  %  ruzne(X,Y) platí pokud za X a Y 
                   %  nejde substituovat tak, aby X a Y byly stejné 

Definovat  takový   predikát  je  pro   nás  zatím  obtížné. Pravděpodobně se vám nelíbil ani popis jeho významu. 
Problém není kupodivu  příliš jednoduchý. 

  Předpokládejme, že  takový predikát máme k dispozici. Pak není žádný problém "vyrobit si sestru". 

sestra(Ses,X):- rodic(Y,Ses),rodic(Y,X),zena(Ses),ruzne(X,Ses). 

     Všechny definice nových predikátů,  se kterými jsme se dosud setkali byly tvořeny jedinou klauzulí. To je však 
spíše výjimečný případ. 

     Definujme  predikát vmanzelstvi(X,Y),  který o  dané dvojici bude platit,  když X a Y  budou manželé 
(nepožadujeme, aby  X byl muž a Y žena). Příslušná definice může vypadat např. takto: 

vmanzelstvi(X,Y):- manz(X,Y). %  Je-li X manželem Y, jsou X a Y  v manželství 
vmanzelstvi(X,Y):- manz(Y,X). %  Je-li Y manželem X, jsou X a Y  v manželství 

Totéž  by šlo  napsat  si  pomocí o p e r á t o r u   s t ř e d n í k , který znamená d i s j u n k c i  - logickou 
spojku  n e b o , i v jedné klauzuli: 

vmanzelstvi(X,Y):- manz(X,Y) ; manz(Y,X). 
  %  Je-li X manželem Y nebo Y manželem X, jsou X a Y v manželství 

Nyní je  již také jasné,  proč příkazem pro  hledání dalšího alternativního řešení byl právě středník. 

Z  důvodů přehlednosti  programů je,  jak uvidíte,  většinou lepší používat tvar s dvěma  klauzulemi, než se 
středníkem. Proto mu, alespoň dokud nezískáte větší  praxi a s ní i vlastní názor, dávejte přednost. 

Pokud  však přesto  chcete se  středníky experimentovat  již teď, vězte,  že 

 operátor čárka  (konjunkce)   v á ž e  v í c  než operátor středník  (d isjunkce) .  

Například 

 p(X):- a(X,Y),b(Y);c(Y),d(X,Y). 

je ekvivalentní s 

p(X):- a(X,Y),b(Y). 
p(X):- c(Y),d(X,Y). 

a ne s 

p(X):- a(X,Y),b(Y),d(X,Y). 
p(X):- a(X,Y),c(Y),d(X,Y). 

Kdybychom chtěli předchozí  definici zapsat pomocí středníku v jedné klauzuli, musíme použít závorky: 

p(X):- a(X,Y),(b(Y);c(Y)),d(X,Y). 

Přehlednost takových zápisů  můžeme zvýšit vhodnou grafickou úpravou, např. 

p(X):- a(X,Y), 
      (b(Y);c(Y)), 
  d(X,Y). 

Výhodou zápisu se středníkem,  ve srovnání se zápisem pomocí dvou  klauzulí,  v  tomto  případě  je,  že  se  při 
navracení po neúspěchu  při snaze  o splnění  predikátu b(Y)  , nemusíme znovu pokoušet  o  splnění  predikátu  a(X,Y),  
ale  rovnou  se snažíme o splnění alternativního  predikátu c(Y). Tato  výhoda může nabýt na váze především, je-li 
vyhodnocení predikátu a časově náročné. 

Chceme-li  tuto  výhodu  zachovat  a  vyhnout  se  používání středníků  ve složitějších  klauzulích, můžeme  si 
vzpomenout  na staré dobré strukturované programování  a jeho zásadu "Rozděl, co můžeš" , a definovat nový predikát: 

bneboc(Y):- b(Y). 
bneboc(Y):- c(Y). 
p(X):- a(X,Y),bneboc(Y),d(X,Y). 

Když  už jsme  se tak  zapletli do  "stylistických" diskusí, nesmíme vám zamlčet ještě jeden pojem, který by se 
vám mohl hodit i v  následujícím  cvičení,  tzv.   a n o n y m n í   p r o m ě n n o u .  

Chceme-li vytvořit  predikát  zenaty(X),  který  by  byl  pravdivý  pro  ty objekty, které mají manželku, můžeme to 
udělat takto: 

zenaty(X):-  manz(X,Y). 

Predikát  zenaty(X)  bude,  jak   již  víme,  splněn,  pokud nalezneme  takovou substituci  za proměnné   X a  Y, po  
níž bude platit manz(X,Y). Užití predikátu ukazuje následující konverzace: 

?-zenaty(X). 
     X = adam ; 
     X = karel ; 
     X = zibrid ; 
     X = jan ; 
     X = jose ; 
     no 

Vidíme tedy,  že v předchozí definici  predikátu zenaty mají proměnné  X  a  Y,  použité  v predikátu  manz  na  
pravé straně pravidla,  rozdílnou  roli.  Proměnná  X  se  vyskytuje  i  jinde v klauzuli  (v tomto  případě  vlevo),  
zatímco proměnná  Y nikde jinde  v klauzuli  není. Nezáleží  tedy vůbec na  jménu této  proměnné.  

Definice 
                     zenaty(X):-  manz(X,Z). 

je  zcela ekvivalentní  definici původní.  Abychom to,  že nám na hodnotě  druhé  proměnné  nezáleží,  vyjádřili 
explicitně, píšeme takovou definici obvykle ve tvaru 

                     zenaty(X):-  manz(X,_). 

Znak podtržítko '_' označuje tzv. anonymní proměnou. Použití anonymní proměnné demonstruje i následující 
konverzace: 

?-manz(On,_). 
     On = adam  ; 
     On = karel  ; 
     On = zibrid  ; 
     On = jan  ; 
     On = jose  ; 
     no 

Srovnejte   získané   odpovědi    s   odpověďmi   na   dotaz ?-manz(On,Ona),  které  jsme  uvedli  výše.  Vidíme,  
že  p o u ž i t í  a n o n y m n í   p r o m ě n n é  v  dotazu mělo  za následek  p o t l a č e n í  v ý s t u p u  h o d n o t  
s u b s t i t u o v a n ý c h  z a  t u t o  p r o m ě n n o u .  

A n o n y m n í  p r o m ě n n é  jsou dalším  prostředkem, který nám pomáhá p s á t  p ř e h l e d n ě j š í  p r o g r a m y . Mnoho 
překladačů prologu vás pomocí varování upozorňuje na skutečnost, že jste některou z proměnných použili v nějaké 
klauzuli jen jednou. Jak jsme již viděli není to samo o sobě chyba, ale velmi často se tak projeví chyby vzniklé 
překlepem. Patř í  k dobrému stylu užívat  na takových místech anonymní proměnnou.   

Cvičení: 

1. Sestavte predikáty 

 tchyne(Tch,X)    %  Tch je tchýní X 
   manzdite(On,Ona,Dite) %  Dite je manželským dítětem 
                      %  dvojice On a Ona 
   vnuk(StRodic,Vnuk)   %  Vnuk je vnukem StRodice 

 2. Sestavte  predikáty  vyjadřující  další  příbuzenské  vztahy (např. švagr,  újec, bratranec, zeť,  dědeček 
z matčiny strany, manžel vnučky, prababička manželovy tety, a pod.). 

3. Tvar prologovského programu 

Nebudeme definovat  tvar programu v  Prologu příliš formálně ze dvou důvodů.  Za prvé by nás to odvedlo  příliš 
daleko k čistě technickým  otázkám, které  pro nás  nejsou tak  podstatné, a  za druhé  není v  Prologu tak   velký rozdíl  
mezi daty  a programem - a  jazyk  sám  obsahuje   poměrně  bohaté  prostředky,  kterými programátor může "měnit 
tvar" programu. 

Jak  jsme  si  již  řekli,  Prolog  "počítá" nikoli s obsahy proměnných   (proměnné  jsou   v procedurálních   
programovacích jazycích  de facto  jména pro  místa v  paměti), ale  s "nápisy". Universální datovou strukturou, se 
kterou Prolog pracuje, je tzv. t e r m -  pojem převzatý z formální logiky.  

Klasifikace  termů  v  Prologu   je  vidět  z  následujícího obrázku: 

term

jednoduchý term struktura

konstanta proměnná

atom číslo

 
obr. 1  Datové typy Prologu 

A t o m y  mohou být v Prologu konstruovány jedním ze tří způsobů: 
�� buď z a č í n a j í   m a l ý m   p í s m e n e m   a  o b s a h u j í  p o u z e  p í s m e n a ,  č í s l i c e  a  z n a k  

 p o d t r h á v á t k o , tedy např. 
rodic   emil   novy_prvek    eMIL   atom1 

�� nebo se s k l á d a j í  p o u z e  z e  s p e c i á l n í c h  z n a k ů , např. 
,      ;      <       -->     <===>     :-      ?-      !      . 

�� nebo jde o z n a k o v é  ř e t ě z c e  uzavřené mezi znaky apostrof (v některých implementacích uvozovky) 
'Adam'   'Horní Počernice'  'kykyryký'         . 

Č í s l a :   V Prologu  máme vždy   k dispozici  celá čísla,  někdy však nejdou zpbrazovat příliš velká čísla ("maxint" je 
malé).   Komerční  implementace  Prologu  obsahují většinou i reálná čísla. 

P r o m ě n n é :  Jak  jsme si  již řekli,  poznají se  tak, že  z a č í n a j í  v e l k ý m  p í s m e n e m  n e b o  z n a k e m  
p o d t r h o v á t k o ,  n e s m ě j í  v š a k  o b s a h o v a t  s p e c i á l n í  z n a k y , . např. 

X   Otec    _    Seznam    _XXX 

S t r u k t u r y  jsou d e f i n o v á n y  r e k u r s i v n ě : 

Struktura je tvořena f u n k t o r e m   (který má danou a r i t u  tj. počet argumentů) a příslušným  počtem 
termů, které jsou  j e h o  a r g u m e n t y . Např. 

rodic(adam,X)  funktor rodic má aritu 2 
datum(23,8,1993)  funktor datum má aritu 3 
okamzik(datum(23,8,1993),cas(21,35)) funktor okamžik má aritu 2 
  funktor cas má aritu 2 

strukturou je i klauzule 

deda(X,Y) :- rodic(X,Z),rodic(Z,Y).  předdefinovaný funktor „:-„ je binární a píše se jako infixový 
operátor 

Není syntaktickou  chybou,  pokud   se  v  jednom  programu vyskytují dva stejně pojmenované funktory 
s různými aritami. Tedy např. klauzule 

ppp(X,Y,Z):- ppp(X,Y) , ppp(Y,Z). 

je naprosto přípustná  a atom ppp v ní  označuje d v a  r ů z n é  f u n k t o r y  (jeden binární a druhý ternální), které 
oba mají jméno ppp. 

Víme již, že p r o g r a m  v  Prologu j e  p o s l o u p n o s t  k l a u z u l í  a že každá klauzule je ukončena 
tečkou.  Klauzule mohou být jednoho ze dvou typů 

fakta    např.     manz(emil,rebeka). 
pravidla     např.  otec(Ot,Dite):- rodic(Ot,Dite),muz(Ot). 

L e v é  s t r a n ě   pravidla říkáme h l a v a  p r a v i d l a ,  p r a v é   pak t ě l o  p r a v i d l a  ( hlava 
je vlastně  prvním argumentem funktoru  :-  a tělo jeho druhým argumentem). 

Na  f a k t y   se  můžeme  podívat  také  jako na speciální typ pravidel,  na  jejichž  pravé   straně  stojí  
standardní  nulární predikát  true,  který  je  vždy  platný 

manz(emil,rebeka):- true.  , 
resp.  jako na  p r a v i d l a  s   p r á z d n ý m  t ě l e m .  Proto můžeme  mluvit o hlavě i u faktů. 

Ještě  jeden termín  se nám  bude hodit  - je to p r o c e d u r a . Procedura v prologovském programu je  
posloupnost klauzulí, jejichž hlavy mají stejný funktor  (tj. včetně arity).  Uvnitř této posloupnosti  je zachováno pořadí 
klauzulí z programu. 

Jak  uvidíme,  v  prologovském  programu  vůbec  n e z á l e ž í  n a  p o ř a d í   k l a u z u l í ,  k t e r é   n e p a t ř í   
d o   s t e j n é  p r o c e d u r y .   Proto je běžné  psát   klauzule  patřící  k  téže   proceduře  v  programu pohromadě. 

Ještě něco  o   k o m e n t á ř í c h .  Již umíme  psát ř á d k o v é  k o m e n t á ř e  uvozené z n a k e m  
p r o c e n t o   %   - text  od tohoto znaku  počínaje až do  konce řádku se bere jako komentář. Někdy  je však opakování 
%  na každém řádku  velmi nepříjemné.  Proto  můžeme  používat i  k o m e n t á ř o v ý c h  z á v o r e k    /*    a  */  . Text 
mezi "závorkami   /*  a   */   je celý chápán jako komentář . Tyto komentáře mohou tedy obsahovat i více řádek, 
resp. mohou končit před koncem řádku: 

 
/* 
     Program našich  klepen (obohacený o  některé nové procedury) napsaný  
tak, že  klauzule patřící  k jednotlivým procedurám  jsou  pohromadě.  
Uvnitř  jednotlivých  procedur jsme pořadí klauzulí nezměnili. 
     Stejně jako v Pascalu nezáleží na umístění klauzulí na řádce ani na 
počtu mezer,  které jednotlivé prvky jazyka oddělují. Programy v Prologu se 
tedy  píší v tzv. volném formátu, jako dnes už prakticky u všech 
programovacích jazyků. 
*/ 

%  Procedura muz 
muz(adam).   muz(karel).  muz(emil). 
muz(zibrid).  muz(jan). muz(jose). 

%  Procedura zena 
zena(lucie).  zena(eva).     zena(helena). 
zena(klara).    zena(kunhuta).   zena(katka). 
zena(ruzena). 

%  Procedura manz 
manz(adam,eva).      manz(karel,helena). 
manz(zibrid,kunhuta).    manz(jan,lucie). 
manz(jose,katka). 

%  Procedura rodic 
rodic(adam,josef).  rodic(adam,hugo). 
rodic(eva,josef).   rodic(eva,hugo). 
rodic(karel,zuzana). rodic(helena,zuzana). 
rodic(helena,alfred).  rodic(karel,alfred). 
rodic(klara,emil).  rodic(karel,emil). 
rodic(kunhuta,eva).    rodic(kunhuta,katka). 

rodic(zibrid,eva).     rodic(zibrid,katka). 
rodic(ruzena,jan).      rodic(jan,adam). 
rodic(lucie,adam). 

%  Procedura otec 
otec(Otec,Dite):- rodic(Otec,Dite),muz(Otec). 

%  Procedura matka 
matka(Matka,Dite):- rodic(Matka,Dite), zena(Matka). 

%  Procedura manzdite 
manzdite(X):- otec(Ot,X),matka(Mat,X),manz(Ot,Mat). 

%  Procedura deda 
deda(D,V):- rodic(R,V),otec(D,R). 

%  Procedura vmanzelstvi 
vmanzelstvi(X,Y):- manz(X,Y). 
vmanzelstvi(X,Y):- manz(Y,X). 

4. Unifikace a backtracking - základ interpretace 
 prologovských programů 

V této kapitole se seznámíme  podrobněji s tím, j a k  " v l a s t n ě  P r o l o g   p o č í t á " .  Již  víme,  že  interpret 
prologu  hledá metodou backtrackingu v  programu vhodnou klauzuli a  substituci tak, aby mohl  "uspokojit dotaz". 
S takovým  mlhavým  a   neurčitým  popisem  jsme  se  mohli spokojit,  dokud "o  nic nešlo",  ale brzy bychom se  
dostali do úzkých. 

Protože idea backtrackingu  (úplného prohledávání do hloubky s návratem) je  vám z programování  přece jen 
známa,  podíváme se nejdříve na problematiku hledání vhodné substituce. 

4.1. Unifikace 
Začneme příkladem. Mějme    velmi   jednoduchý    (až   prostoduchý)    program reprezentující  naše znalosti  o 

směru  úseček. Bude  pracovat se čtyřmi predikáty: 

bod(X,Y)           binární  predikát  bod  reprezentuje bod v rovině, jeho argumenty jsou 
 souřadnice příslušného bodu , 
usecka(Zac,Kon)  binární  predikát   usečka  reprezentuje úsečku,  argumenty  jsou  její 
 počáteční a koncový bod , 
vodorovna(Usecka) unární predikát vodorovna používáme tak, že jeho argumentem je úsečka; 
 o ní pak tvrdí,  že  má  vodorovný  směr  (tj. je rovnoběžná s osou x) , 
svisla(Usecka)     unární predikát svisla používáme tak, že má za  argument úsečku; o  té pak 
  predikát tvrdí, že  má  svislý  směr  (tj. je rovnoběžná s osou y) . 

Zkuste nejprve příslušný program sestavit samostatně ! 

svisla( usecka( bod(X,_),bod(X,_) ) ). 
   %  úsečka je svisla, pokud oba její koncové 
   %  body mají stejné X-ové souřadnice 
vodorovna( usecka( bod(_,Y),bod(_,Y) ) ). 
   %  úsečka je vodorovná, pokud oba její koncové 
   %  body mají stejné Y-ové souřadnice 

     Pokusme  se z  tohoto programu  "vydolovat, co  se dá". Jaké dotazy vlastně můžeme tomuto programu položit? 

?- vodorovna(usecka(bod(1,3),bod(2,4))). %  Je úsečka s krajními body (1,3)  
  no  %  a (2,4) vodorovná ? 
?- svisla(usecka(bod(2,3),bod(2,4))).  %  Je úsečka s krajními body (2,3) 
   yes  %  a (2,4) svislá ? 
?- svisla(usecka(bod(1,3),bod(X,4))).  %  Jaká  musí  být  x-ová 
  X = 1 %  souřadnice bodu, jehož  y-ová souřadnice je 
   %   rovna 4, aby  úsečka, která  spojuje tento 
          %  s bodem (1,3) byla svislá ? 

To nebylo nic nového, jakou  odpověď byste však očekávali na dotaz 

?- svisla(usecka(bod(1,Y1),Z)).     ?  (1) 

 Je jasné,  že za proměnnou  Z může být  substituována jedině struktura s funktorem bod. Jaké však mají být její 
argumenty ? Co byste řekli takovéto odpovědi 

 Y1=7   Z=bod(1,10087)    ?  (2) 

Tato  odpověď by  byla  sice  "správná" (vždyť  přece úsečka spojující body (1,7) a (1,10087) je svislá), ale současně 
i v e l m i  p o d e z ř e l á . Na  základě čeho Prolog  stanovil substituce za  y-ové souřadnice obou bodů ? 

Jak  jsme  si  řekli,  "neví"  Prolog  o  predikátech našeho programu nic jiného než to, co jsme mu o nich 
v programu sdělili. Nemůže  tedy vědět,  že za  druhý argument  predikátu bod se mají dosazovat čísla, - a i  kdyby to 
věděl,  jak by přišel  zrovna na hodnoty 7 a 10087 ? 

Navíc, uvědomíme-li si skutečný význam dotazu (1), zjistíme, že odpověď (2) vlastně vůbec  správná nebyla. 
Dotazem (1) jsme se totiž ptali: 

"V jakých bodech mohou končit svislé úsečky, které začínají v bodě s x-ovou souřadnicí rovnou 1 ?". 

A správná odpověď na tuto otázku je: 

 "V libovolném bodě, jehož x-ová souřadnice je rovna 1." (3) 

Přesně vzato  ani tato odpověď správná  není, neboť, aby šlo o skutečnou úsečku,  musí se  její koncový  bod lišit  
od počátečního,  ale tento poznatek  jsme do programu  nezabudovali (a ani  to, zatím, neumíme). 

Požadovanou odpověď  (3)  však  Prolog  může  nalézt  - narozdíl od nepochopitelné odpovědi (2)  -  velmi 
snadno. S t a č í ,  a b y  n e s u b s t i t u o v a l   z b y t e č n ě ,  k d y ž   n e m u s í . Odpověď  na dotaz  (1) by tedy mohla 
vypadat takto: 

 Y1 = cokoliv 
 Z  = bod(1,Y2)       (4) 
 Y2 = cokoliv         , 

kde cokoliv by byl předdefinovaný atom, který by nám symbolizoval skutečnost,  že  na  hodnotě  dané  proměnné  
splnění daného cíle nezáleží.  To bychom  však zaváděli  zbytečně nový  - a nešikovný - formalismus.  Zadáte-li  dotaz  
(1)  vašemu  interpretu Prologu dostanete odpověď v tvaru podobném následujícímu: 

 Y1 =  _323     Z = bod(1,_324)      (5) 

Znakem  podtrhovátko '_'  začínají, jak  již víme, jména proměnných. Interpret  většinou  generuje   jména   
i n t e r n í c h   p r o m ě n n ý c h   připojením přirozeného čísla za tento znak. 

Jaký je z toho všeho závěr ? 

I n t e r p r e t    P r o l o g u    s e    s n a ž í    n a l é z t   n e j o b e c n ě j š í   s u b s t i t u c i ,  k t e r á  
s p l n í  d a n ý  c í l , tzn.  n e s u b s t i t u u j e   z b y t e č n ě ,  p o k u d  n e m u s í . 

 O d p o v ěď ,  ve které se v y s k y t u j e  p r o m ě n n á , znamená, že n a   d a n é  h o d n o t ě  n e z á l e ž í  (a může 
být za ní dále substituováno cokoliv). 

Uveďme ještě jeden příklad dotazu pro náš program 

 ?- horiz(usecka(bod(1,Y1),bod(X,Y2))).             (6) 

V tomto případě odpověď může být např. takováto: 

 Y1 = _432    X = _433    Y2 = _432    .   (7) 

Všimněte si, že z a  p r o m ě n n é  Y 1  a  Y 2  b y l a  s u b s t i t u o v á n a  t a t á ž  i n t e r n í   p r o m ěn n á .  

Došlo  tedy  k e   z t o t o ž n ě n í   p r o m ě n n ý c h   Y 1   a   Y 2  z dotazu. Bude-li  se s  odpovědí 
dále  pracovat, musí  být za obě dosazeno přesně totéž. 

Abychom mohli  to, co nyní  již velmi pravděpodobně  správně tušíte, formulovat přesněji, potřebujeme zavést dva 
pojmy. 

 Musíme definovat co to znamená, že si d v a  t e r m y  o d p o v í d a j í  (match) a 
 jak probíhá proces u n i f i k a c e , který to demonstruje : 

  1. Jestliže T 1  i  T 2  j s o u  p r o měn n é ,  pak si odpovídají a výsledkem jejich unifikace je ztotožnění těchto 
proměnných. 

  2.  Jestliže  T 1   j e   p r o měn n á   a   T 2   j a k ý k o l i   t e r m   (různý  od proměnné), pak si termy T1  a T2 
odpovídají a výsledkem  jejich unifikace je substituce termu T2 za proměnnou T1 

T1 <- T2  . 

  3.  Jestliže T 2   j e   p r o měn n á   a   T 1   j a k ý k o l i   t e r m  (různý  od  proměnné), pak si termy T1  a T2  
odpovídají a výsledkem jejich unifikace je substituce termu T1 za proměnnou T2 

T2 <- T1  . 

  4. Jestliže jsou  T1 i  T2 jednoduché termy,   různé od proměnných ,   pak  si odpovídají  jen  k d y ž   
j s o u  i d e n t i c k é   a jejich  unifikace je prázdná akce. 

  5. Jestliže  T1  i  T2  jsou struktury ,  pak si  odpovídají, pokud  j s o u   t v o ř e n y   t ý m ž   f u n k t o r e m  
(z definice pak musí být jejich arita stejná) a  pokud si j e j i c h   p a r a m e t r y  n a v z á j e m  
o d p o v í d a j í  (první  prvnímu, druhý druhému, ... , atd.) a jejich unifikace spočívá v unifikaci odovídajích si 
dvojic parametrů. 

  6. V  ž á d n ý c h   j i n ý c h  p ř í p a d e c h  s i  d v a  t e r m y   n e o d p o v í d a j í  a nelze je unifikovat. 

Uveďme několik příkladů unifikace termů: 

a) termy 
  f(t(a,X),t(B,b),Y)                       a 
  f(t(A,b),t(f(t(a,a),t(z,b),Z),b),r) 

  si odpovídají a unifikací získáme substituce: 

     X <- b  ,   A <- a  ,  B <- f(t(a,a),t(z,b),Z)  a   Y <- r  . 

b) termy 
  g(A,B,a(A,a(x,B)))                       a 
  g(v,t,a(A,a(x,w))) 

 si neodpovídají (nejde je unifikovat): 
�� unifikací prvního argumentu dostáváme substituci A <- v, 
�� unifikací druhého  B <- t, 
�� funktor třetího argumentu v obou termech je a, 
�� jeho první argumenty si odpovídají, 
�� druhé argumenty mají (po provedení již vynucené substituce B <- t ) tvar: 

                 a(x,t)     a      a(x,w)       a tedy si neodpovídají. 

verze 3.03   strana 14  



 

Poznámka. 

S daným  algoritmem zjišťování, zda si  dva termy odpovídají jsou  i (spíše  teoretické) problémy.  Zkusme 
zjišťovat,  zda lze unifikovat termy 
  X     a       f(X)   . 
Námi popsaný algoritmus se zacyklí, hledaje substituci tvaru 
  X=f(f(f(f( ........  ))))     . 
I když my na první pohled  vidíme, že vhodná (konečná) substituce nemůže existovat. 

Konkrétní  implementace  Prologu,  i  když používají popsaný algoritmus, mají takové případy ošetřeny, 
  aby zbytečně nepadaly. 

Zdůrazněme na závěr tohoto odstavce ještě několik důležitých skutečností  týkajících  se  proměnných  
v Prologu.  (Abychom si rozuměli, budeme  se - pro  jistotu - vyjadřovat  "velmi lidově". Prosím tedy šťouraly, aby mne 
nyní nechytali za slovo): 

1. P r o měn n é   v  Prologu  j s o u   " l o k á l n í  v  k l a u z u l i ,  v  n í ž  j s o u  p o u ž i t y " .  
To znamená, že je-li v různých klauzulích použito stejného označení proměnné, nemají spolu tyto proměnné 
(z hlediska programu) vůbec nic společného. 

2. Proměnné v Prologu  n e o z n a č u j í  m í s t o  v  p a m ě t i  p o č í t a č e  (jako je tomu v procedurálních 
jazycích  ala Pascal), a l e  " j e n "  m í s t o  v  t e r m u ,   k a m   l z e   s u b s t i t u o v a t  j i n ý  t e r m . To, 
že  se  v  klauzuli vyskytuje tatáž vícekrát, znamená,  že na příslušná místa je nutno substituovat totéž. 

3 .  V Prologu tedy n e e x i s t u j í  " g l o b á l n í  p r o měn n é ", do nichž bychom si mohli uložit nějakou 
hodnotu a někdy později ji odtud znovu "přečíst" pro další použití. V e š k e r ý  p ř e n o s  i n f o r m a c í  
p r o b í h á  p ř e d á v á n í m  p a r a m e t rů .  

 Nyní již můžeme poměrně přesně popsat algoritmus interpretace prologovských programů. 

4.2. Algoritmus interpretace prologovských programů 
Jak jsme si již řekli, je základem interpretace prologovských programů algoritmus backtrackingu, tj. prohledávání 

do hloubky s návratem v případě neúspěchu. Nebudeme jej popisovat přesně,  zbytečně  by  to  náš  text  prodloužilo  
a navíc to lze ponechat čtenáři jako pěkné cvičení (z procedurálního programování). 

Připomeňme  si  velmi  volně  základní  ideu  prohledávání s návratem: 

J d i   k u p ř e d u ,  d o k u d   mů ž e š ,  a   p a m a t u j  s i   v š e c h n a  r o z h o d n u t í ,  k t e r á  
p ř i  t o m  d ě l á š .  P o k u d  j i ž  n e mů ž e š  d á l ,  v r a ť   s e  n a   m í s t o  p o s l e d n í h o   
r o z h o d n u t í ,  r o z h o d n i   s e  j i n a k ,  a  p o k r a ču j  s t e j n ý m  z pů s o b e m  d á l .  

Pokud prostor,  kterým procházíme, je konečný  a nemá cykly, pak  popsaným  algoritmem  projdeme  celou  jeho 
část, která je dostupná z místa našeho startu. 

Z  programátorského hlediska se algoritmus backrackingu realizuje pomocí zásobníku, ve kterém si pamatujeme 
cestu i rozhodování na jejích jednotlivých místech.  

Činnost  interpretu  prologovských  programů  lze plně popsat pomocí rekursivní procedury splňování cíle. 
Cílem přitom bude  term, který je buď dotazem  nebo vznikl z některého predikátu zapsaného v  programu substitucí za 
(některé) proměnné v něm obsažené. Přesné zformulování ponecháme čtenáři jako cvičení a spokojíme se s několika 
poznámkami: 

1. P r v n í m  c í l e m , který se interpret snaží splnit j e  d o t a z  zadaný uživatelem. 
2. P ř i  s p l ňo v á n í  c í l e  interpret h l e d á  p r v n í  k l a u z u l i ,  j e j í ž  h l a v a  o d p o v í d á  t o m u t o  c í l i .  

Při  unififikaci  cíle s hlavou této klauzule se provedou  jisté  substituce  (nejen v  hlavě, ale v celé klauzuli).  
Do zásobníku se uloží číslo použité klauzule a substituce, které byly provedeny. 

3. Klauzule, jejíž  hlava odpovídá právě splňovanému cíli se při cestě výpočtu "kupředu" hledá vždy od počátku 
programu. 

4. Pokud je nalezená o d p o v í d a j í c í  k l a u z u l e  f a k t e m ,  c í l  u s p ě l . 
5. Pokud je to  p r a v i d l o , pak prolog  přejde k p o s t u p n é m u  s p l ňo v á n í  c í l ů  v z n i k l ý c h  

p r o v e d e n ý m i  s u b s t i t u c e m i  z  p r e d i k á t ů  t ě l a  t é t o  k l a u z u l e .  Pokud u s p ě j í  v š e c h n y  
c í l e  t ě l a  této klauzule, u s p ě l  i  p ův o d n í  c í l .  (Při této "cestě kupředu" se zásobník naplňuje). 

6. Pokud  s e  n ě k t e r ý  c í l  n e p o d a ř í  s p l n i t  -  d o j d e  v souhlase se strategií backtrackingu  
k  n a v r a c e n í  v ý p o č t u . Při  tom se ze zásobníku vyzvedne poslední položka, prolog odvolá naposledy 
provedené substituce a snaží  se najít další klauzuli, jejíž hlava odpovídá splňovanému cíli. Při  tom se  
program pochopitelně  prohledává nikoli od začátku, ale od aktuální pozice v programu (která byla právě  
vyzvednuta  ze zásobníku). Pokud tak dojdeme až na konec programu, tento cíl definitivně nejde splnit 
(pochopitelně však může být později splněn jiný cíl odvozený jinou substitucí z téhož predikátu v programu) 
a  navracení  pokračuje dále stejným způsobem. 

4.3. Krabičkový model výpočtu, princip ladění 
Pro znázornění  průběhu vyhodnocování prologovských  cílů se často  používá   tzv.  k r a b i č k o v ý  

m o d e l   v ý p o č t u   prologovských programů. 
Každému  v o l á n í    p r e d i k á t u   o d p o v í d á   k r a b i č k a    s e   č t y řm i  b r á n a m i . Dvě jsou vstupní a 

dvě výstupní. 
 

 

REDO 

EXIT 

FAIL 

CALL 

Horní  dvě  brány  CALL   a   EXIT   se používají  p ř i  p o s t u p u  v ý p o č t u  v p ř e d . 

Vstupní  bránou CALL  vstupuje výpočet  do splňování  daného cíle.  Podařilo-li  se  daný  cíl  splnit  (tj.  
byla-li nalezena substituce,  která  převedla  daný  term  na  takový,  který  jde z programu odvodit), odchází výpočet  
branou EXIT k dalšímu cíli, který se má splnit jako další. 

Spodní dvě brány  REDO  a   FAIL   se  použijí, když s e  v ý p o č e t  n a v r a c í . 

Pokud se následující cíl nepodařilo splnit, vrací se výpočet z brány FAIL tohoto  cíle zpět do našeho branou 
REDO.  To je signálem, že  se  má  hledat  další  alternativní  substituce  pro  splnění aktuálního  cíle. Podaří-li  se 
to,  výpočet odchází  branou EXIT znovu k následující  krabičce, a když se to  nepodaří, výpočet se vrací zpět 
branou FAIL k předcházející krabičce. 

Na představě  tohoto modelu je  velmi často založen  způsob, kterým se prologovské programy obvykle ladí. 

Programátor může předepsat, k t e r é  p r e d i k á t y  c h c e   při výpočtu v ladicím  režimu  s l e d o v a t   (obvykle  
předdefinovaným predikátem   s p y  ). Je-li z a p n u t  l a d i c í  r e ž i m  (obvykle  předdefinovaným predikátem 
t r a c e ,  v y p í n á  j e j  predikát  n o t r a c e ) v ý p o č e t  s e  vždy z a s t a v u j e  p ř i  p r ů c h o d u  b r a n o u   
krabičky odpovídající některému ze sledovaných predikátů. 

P ř i   každém   takovém  z a s t a v e n í  s i  mů ž e   programátor  na obrazovce p ř e č í s t  a k t u á l n í  s t a v  
t e r m u ,  který se  program právě pokouší splnit resp. byl právě  splněn. 

Navíc mů ž e  o v l i v n i t  d a l š í  p r ůb ě h  v ý p o č t u  (např. zrušení ladicího  režimu, přidání  dalšího predikátu  
ke sledovaným resp. zrušení sledování jiného, průchod  přes některé brány sledovaných predikátů bez zastavení,  
vyvolání pomocného výpočtu prologovským dotazem - po obdržení odpovědi pokračuje sledovaný výpočet, provedení 
libovolného příkazu operačního systému atp.). 

5. Význam prologovských programů. Rekurse. 

Tato kapitola  bude pro vás pravděpodobně  o  n ě c o  o b t í ž n ě j š í  n e ž  d o s a v a d n í  t e x t .   Je  však   
klíčová jak pro praktické programování v Prologu, tak i  pro pochopení jeho principů a role mezi ostatními 
programovacími jazyky. 

Proto je dobré se k ní občas vrátit. Pak možná objevíte nové významy a další látku k přemýšlení. 

5.1.  Deklarativní význam prologovského programu 

Prolog je praktickým pokusem o realizaci myšlenky l o g i c k é h o  p r o g r a m o v á n í , které se snaží   n a h r a d i t   
k l a s i c k é   p r o g r a m y  předepisující přesný algoritmus řešení úlohy jejím p o p i s e m  p o m o c í  p r o s t ř e d ků  
m a t e m a t i c k é  l o g i k y .  

Každému p r o g r a m u  odpovídá jednoznačně f o r m u l e  v ý r o k o v é h o  p o č t u ,  k t e r á  
u r č u j e  j e h o  (deklarativní1) v ý z n a m . Tuto formuli budeme definovat postupně. Nejprve budeme pro 
každou klauzuli definovat formuli, která udává význam této klauzule. V ý z n a m  p r o g r a m u  je pak definován 
jako k o n j u k n c e  v ý z n a m ů  j e h o  k l a u z u l í . 

Každé klauzuli prologovského  programu  přiřadíme formuli výrokového počtu,   která v y j a d ř u j e  
v ý z n a m  této klazule následujícím způsobem: 

fakt  p  znamená platnost   tvrzení  p   
a   
pravidlo  p :-  r1 ,r2 ,r3 , ... ,rn . 
  má význam implikace r1 & r2 & r3 &  ... & rn => p,  

 která tvrdí ,  že j sou-l i  p latná všechna tvrzení  r1 ,r2 ,r3 , . . . , rn ,  
 j e  platné i  tvrzení  p .  

Pokud   ta to  formule  obsahuje proměnné,   znamená tvrzení  o existenci  objektů ,  j e j ichž 
subst ituování   do formule j i  p řevede na platné tvrzení  o těchto objektech .  

Prologovský  program  je  vlastně  množinou  klauzulí,  j e h o  v ý z n a m  j e  d á n  k o n j u n k c í  v š e c h  
f o r m u l í ,  k t e r é   u r ču j í  v ý z n a m  j e d n o t l i v ý c h  k l a u z u l í  p r o g r a m u . Tomuto významu se říká 
d e k l a r a t i v n í  v ý z n a m  p r o g r a m u .  

Než  pokročíme  dále,  uvědomme  si,  že d e k l a r a t i v n í  v ý z n a m  n e z á v i s í   a n i   n a   
p o ř a d í   k l a u z u l í   v   p r o g r a m u ,   a n i   n a  p o ř a d í  p r e d i k á t ů  v  t ě l e  
j e d n o t l i v ý c h  p r a v i d e l .  Proto  jsme také  mohli  na  předchozích řádcích  o programu mluvit jako o 
množině  klauzulí , a nikoli o posloupnosti klauzulí, jako dosud. 

Skutečnost, že programovací jazyk může mít takovou vlastnost, je jistě fascinující. Jenže,  jak za chvíli uvidíme, 
tak jednoduché to zase není. Pouze deklarativní správnost programu totiž nestačí k tomu, aby "počítal správně". 

P lat í  však dů leži té  tvrzení:  

  J e - l i   p r o g r a m   d e k l a r a t i v n ě   s p r á v n ý ,   n e m ů ž e m e   s e   o d  
n ě j  d o č k a t  š p a t n é h o  v ý s l e d k u .  N e m u s í m e   s e  v š a k  n u t n ě  v ý s l e d k u  d o č k a t  
vůb e c  -  v ý p o č e t  p r o g r a m u  s e  mů ž e  z a c y k l i t .  

Můžeme si to představit  tak, že  p r o g r a m  " v y m e z u j e  p r o s t o r "  p l a t n ý c h  t v r z e n í  . Vyhodnocování 
dotazů je pak vlastně prohledávání takto vymezeného prostoru.  

Je-li  program   deklarativně  správný,  pak  vymezený prostor  obsahuje  právě  všechny  správné  odpovědi na 
libovolný dotaz. Prolog v něm tedy nemůže nalézt žádnou špatnou odpověď. To však  bohužel  ještě  nezaručuje,  že  
výpočet  při vyhodnocování dotazu v tomto prostoru "nezabloudí" (zacyklí se) a žádné odpovědi se tedy nedočkáme. 

                                                           
1 Přívlastku deklarativní užíváme proto, abychom tento význam programu odlišili od významu procedurálního, na 

který jste zvyklí z jiných programovacích jazyků. 

5.2. Rekursivní definice predikátu predek 
Zkusme definovat predikát 

 predek(Predek,Potomek)    , 

který bude znamenat, že objekt Predek je předkem objektu Potomek. 

Uvědomme  si, že  úloha, se  podstatně liší  od té, kdy jsme definovali např. dědu, předka z  "druhého kolene". Pro 
to stačilo dvakrát  použít  na  pravé  straně  pravidla  predikát rodic (pro předka  z třetího  kolene -   pradědečka by  tam 
musel  být rodic třikrát).  Nový  predikát  však   musí  vystihnout  předky  nejen z prvního,   druhého   a    třetího   
kolene,   ale   potenciálně z "nekonečného  počtu"  kolen.  Nevystačíme  tedy  s pravidly, na jejichž pravé straně je jen  
predikát rodic. Musí přijít ke slovu rekurse ( s níž Prolog stojí a padá ). 

Rekursivní  definice předka  vám jistě  nebude dělat potíže. 

 Každý předek objektu  X je buď jeho rodičem,  nebo rodičem jiného předka X. 

%  ============================================================== 
%    predek(?Pred,?Pot)   Pred je předkem potomka Pot 
%  ============================================================== 
predek(Rod,Pot):- rodic(Rod,Pot). %  Rod je rodičem potomka Pot 
predek(Pred,Pot):- 
     rodic(Pred,MlPred),       %  Pred je rodičem mladšího 
     predek(MlPred,Pot).           %  předka Mlpred potomka Pot 
 

Dříve než pokročíme dále, musíme si vysvětlit, co vlastně znamenají otazníky před argumenty predikátu předek 
v „komentářové hlavičce“. Není to součást jazyka (může se tedy vyskytovat jen v komentářích), ale více méně 
všeobecně akceptovaná konvence pro specifikovaní toho, že autor procedury něco předpokládá o dotazech, které budou 
kladeny. Před argument v komentáři můžeme napsat jeden ze tří znaků + , - , ¨? . 

S následujícím významem: 
- „výstupní argument“ - autor komentáře dává najevo, že v dotazech má na odpovídajícím místě stát 

 nekonkretizovaná proměnná, 
+ „vstupní“ parametr – autor komentáře dává najevo, že v dotazech má na odpovídajícím místě stát 

 již konkretizovaný term, 
? argument může být jak vstupní, tak i výstupní tj. v dotazech může na odpovídajícím místě stát 

„cokoli“ . 

Pokud takový znak v komentáři neuvedeme, má to stejný význak jako by tam byl otazník. 

Podle této komentářové konvence jsme tedy v definici predikátu predek dali najevo, že jsme při jeho definici 
neměli na mysli nějaké „specializované“ dotazy. 

Zkusme, jaké  výsledky dá procedura predek,  bude-li doplněna k programu našich klepen z konce kapitoly 3. 

?- predek(lucie,klara).  %  Je Lucie předkem Kláry ? 
     no 
?- predek(jan,X).   %  Kdo je potomkem Jana ? 
     X = adam    ; 
     X = josef   ; 
     X = hugo    ; 
     no 
?- predek(X,hugo).      %  Kdo je předkem Huga ? 
     X = adam    ; 
     X = eva     ; 
     X = kunhuta ; 
     X = zibrid  ; 
     X = ruzena  ; 
     X = jan     ; 

     X = lucie   ; 
     no 

Cvičení 

Neměla by se procedura predek jmenovat raději potomek ? 

Pokud  jste pochopili,  co se  po vás  v předchozím  cvičení vlastně chtělo, udělali jste mi radost. 

V jistém smyslu  se totiž dá říci, že  jméno potomek by bylo pro tuto proceduru o mnoho adekvátnější. Proč ? 
Procedura predek  z j i š ť u j e  t o t i ž  p o t o m k y  daleko p ř i r o z e n ě j i  a  r y c h l e j i  n e ž  p ř e d k y . 

Jak se vyhodnocuje dotaz typu 

 ?- predek(Pred,konkretni_potomek). , 

kde konkretni_potomek je konkrétní atom, pokud nejde o rodiče tohoto potomka ? 

Velmi komicky  - postupně se procházejí  vůbec všechny údaje o dvojicích (Rodič,Dítě) a pak se pro každé takové 
Dítě zjišťuje, zda  náhodou není  předkem vstupního  (nebo raději vstupujícího?) konkrétního potomka. 

Představíte-li si  údaje o populaci několika  tisíc osob, je to spíš námět na další  vtip o policajtech, tentokrát 
hledajících v archivu, než algoritmus vhodný pro řešení dané úlohy. 

Naproti tomu, dotaz typu 

 ?- predek(konkretni_predek,Pot). 

se vyhodnocuje poměrně rozumně.  Nestačí-li nám dítě objektu konkretni_predek, generujeme pro všechny jeho 
děti (těch nemůže být příliš mnoho) jejich potomky - opět napřed jejich děti, atd. 

"Zdravý  selský rozum"  nám napovídá,  že je  lepší když  se nejprve snažíme  splnit cíle, v nichž  alespoň nějaký 
argument je konkretizován. Budou-li tedy převládat dotazy po předcích nad dotazy po potomcích, bylo by lepší 
definovat předka jinak: 

%  ============================================================== 
%           jiná implementace predka vhodnější pro dotazy 
%                     typu pred(-Pred,+Pot) 
%    predek2(?Pred,?Pot)   Pred je předkem potomka Pot 
%  ============================================================== 
predek2(Rod,Pot):- rodic(Rod,Pot) %  Rod je rodičem potomka Pot 
predek2(Pred,Pot):-            %  jeho předky 
     rodic(Rod,Pot), 
     predek2(Pred,Rod).      %  jsou i předci jeho rodičů 

5.3. Procedurální význam programu 
Predikát predek2  byl založen na  (i když z  našeho hlediska jen velmi  nepatrně) odlišné úvaze,  než predikát 

predek. 

  Zkusme však  z k o u m a t  p r e d i k á t y ,  které v z n i k n o u   z  p r e d i k á t u   p r e d e k  j e n  " p e r m u t a c í "   
j e h o  k l a u z u l í   a  c í l ů   v  n i c h .  Dostaneme tři  další možnosti. Prozkoumejme, zda jsou všechny procedurálně 
správné, tj. zda dávají  vždy   správné  výsledky  a   zda  vydají  (po   odmítání středníkem) všechny správné výsledky. 
Predikát predek je správný i procedurálně. Poměrně snadno se o tom  můžete sami  přesvědčit. Jak  to však  bude se  
správností následující jeho modifikace: 

%  ============================================================== 
%     predek3    -     jiná varianta procedury predek 
%                    má stejný deklarativní význam jako predek !! 
%    predek(?Pred,?Pot)   Pred je předkem potomka Pot 
%  ============================================================== 
predek3(Pred,Pot):- 
     predek3(MlPred,Pot), 
     rodic(Pred,MlPred). 

Predek3(Rod,Pot):- 
     rodic(Rod,Pot). 

Cvičení   Jak bude probíhat konverzace s programem po dotazech 

  ?- predek3(jan,X).        ?- predek3(X,hugo). 

Pokud jste uhodli, že to žádná konverzace nebude - prolog se totiž při vyhodnocování těchto  dotazů zacyklí, 
blahopřejeme vám. Je vidět,  že jste dosavadnímu výkladu  perfektně rozuměli. Pokud ne, nic si z toho nedělejte, tohle 
by překvapilo kdekoho. 

Proč se program při pokusu splnit cíl predek3 musí zacyklit bez ohledu na to, které argumenty jsou (a zda jsou) 
konkretizovány? 

Splnění cíle predek3(X,Y) je totiž převedeno na splnění cíle predek3(Z,Y) a  čehosi jiného potom.  Vyhodnocení 
tohoto cíle  si vynucuje vyhodnocení dalšího  cíle predek3(W,Z), atd donekonečna. 

V teorii gramatik se takovým pravidlům, jako je první klauzule predikátu predek3, říká  l e v ě   r e k u r s i v n í  
(definovaný predikát je i prvním cílem vpravo). Pokud je první klauzule nějaké procedury levě rekursivní, pak se  
výpočet této procedury zacyklí vždy.  

Ve skutečnosti to donekonečna nebude, protože interpretu přeteče zásobník a dotaz skončí hláškou podobné této 
                     error - stack overflow         . 

A máme po idyle. Procedura predek3 j e  d e k l a r a t i v n ě  s p r á v n á ,  n e d á  v š a k  n i k d y  
ž á d n ý  v ý s l e d e k . 

Jak to bude s  dalšími dvěma "permutacemi" procedury predek? Začněme tou, která nemění klauzule, jen jejich 
pořadí. 

 
%  ============================================================== 
%               další varianta procedury predek , 
%                také deklarativně ekvivalentní 
%    predek4(?Pred,?Pot)   Pred je předkem potomka Pot 
%  ============================================================== 
predek4(Pred,Pot):- 
     rodic(Pred,MlPred), 
     predek4(MlPred,Pot). 
Predek4(Rod,Pot):- 
     rodic(Rod,Pot). 

Zkusme opět naše dvě otázky. 

?- predek4(jan,X).      %  Kdo je potomkem Jana ? 
     X = josef   ; 
     X = hugo    ; 
     X = adam    ; 
     no 

?- predek4(X,hugo).     %  Kdo je předkem Huga ? 
     X = kunhuta ; 
     X = zibrid  ; 
     X = ruzena  ; 
     X = jan     ; 
     X = lucie   ; 
     X = adam    ; 
     X = eva     ; 
     no 

Vidíme, že jsme dostali stejné výsledky, ale v jiném pořadí.  

Cvičení 

  1. Pokuste  se  charakterizovat  pořadí  v  jakém vydávají řešení procedury predek a predek4 ! Odůvodněte 
svoje tvrzení. 

2. Pokuste  se  dokázat,  že  procedura  predek4  je procedurálně  správná (tj. dá vždy správný výsledek). 

A jak to bude vypadat se čtvrtou modifikací naší procedury ? 

%  ============================================================== 
%               poslední deklarativně ekvivalentní varianta 
%                         procedury predek 
%    predek5(?Pred,?Pot)   Pred je předkem potomka Pot 
%  ============================================================== 
predek5(Rod,Pot):- 
     rodic(Rod,Pot). 
Predek5(Pred,Pot):- 
     predek5(MlPred,Pot), 
     rodic(Pred,MlPred). 

Reakce bude následující: 

?- predek5(jan,X).       %  Kdo je potomkem Jana ? 
     X = adam    ; 
     X = josef   ; 
     X = hugo    ; 
     error - stack overflow 

?- predek5(X,hugo).      %  Kdo je předkem Huga ? 
     X = adam    ; 
     X = eva     ; 
     X = jan     ; 
     X = lucie   ; 
     X = kunhuta ; 
     X = zibrid  ; 
     X = ruzena  ; 
     error - stack overflow 

Program tedy  najde pro tyto dotazy  všechna řešení, ale pak zabloudí - a není již schopen oznámit, že jsou řešení 
všechna. 

Cvičení 

Může se stát, že  procedura predek5 nenalezne všechna řešení pro některý dotaz (případně v jiné 
  "rodině") ? 

 I když se to nezdá, dělají začátečníci v ě t š í   c h y b y   v  d e k l a r a t i v n í m   v ý z n a m u   jimi konstruovaných 
procedur, než v procedurálním. Vždy proto v ž d y  z a č í n e j t e  s  ú v a h a m i  o  t o m ,   z d a  j e  v á š  p r o g r a m   
a l e s p oň   d e k l a r a t i v n ě   s p r á v n ý .  Věříme,  že  vás  tato kapitola neodradila. Pojďme však již programovat! 

6. Seznamy 
Jak jsme si již řekli, pracuje prologovský program s obecnými termy. Programátor si tedy může  (a musí) pomocí 

nich vytvářet vlastní datové struktury sám. Jedinou výjimkou je předdefinovaná datová struktura seznam, která slouží 
k reprezentaci posloupnosti (termů). 

 P r á z d n ý   s e z n a m   je označen  a t o m e m  [   ]  , k reprezentaci neprázdného seznamu slouží binární 
funktor tečka '.' . N e p r á z d n ý  s e z n a m  je tedy tzv.  t e čk a - d v o j i c e  (terminologie pochází z jazyka LISP) 

.(Hlava,Tělo), 

kde  Hlava  je  první  prvek  seznamu  a Tělo je seznam tvořený zbývajícími prvky seznamu. 

Tedy například 
   .(a,.(b,.(c,[ ])))    (1) 

je správný zápis seznamu obsahujícího tři prvky  a,b,c . Tento zápis je však poněkud těžkopádný, proto  je zaveden 
zjednodušený zápis pomocí výčtu prvků v  hranatých závorkách. Uvedený  seznam můžeme tedy (ekvivalentně!) zapsat 
také jako 
   [a,b,c] .    (2) 

     Operátor "svislá čárka" '|' ,  který se používá v tomto typu zápisu, umožňuje vyznačit začátek seznamu. 
Můžeme  tedy ekvivalentně psát také 
   [a|[b,c]] , [a,b|[c]]   .    (3) 

Zápis seznamu s použitím operátoru | má tvar 

[ Začátek | Tělo ] , kde 
Začátek je výčet (ne seznam) prvků stojících na začátku definovaného seznamu 
     a 
Tělo je seznam tvořící  zbytek definovaného seznamu (přitom je-li tento zbytek  prázdný, není třeba jej  
uvádět). Protože Tělo je také seznam, můžeme pro jeho zápis použít tutéž konvenci. 

Pro náš seznam o třech prvcích můžeme tedy kromě výše uvedených zápisů použít i následující (poněkud 
těžkopádné): 

    [a,b,c|[]] , [a,b|[c|[]]] , [a|[b,c|[]]] , [a|[b|[c]]] , 
                 [a|[b|[c|[]]]] , [a|[b,c|[]]] . 

     Všechny tyto  zápisy jsou "zkratkami"  pro zápis (1). 

Pozor  však na  zápisy,  které  vzniknou záměnou  operátoru | a čárky  ! Těmto  chybám, vzniklým  většinou 
přepisem,  se ubrání málokdo. Např. 

[ [a,b] , c] je dvouprvkový seznam o prvcích [a,b] a c 
[ [a,b] | c] není seznam, protože za operátorem | nenásleduje seznam 
[a,b,[c]]   je seznam o třech prvcích  a,b a [c] (jednoprvkový seznam) 
[a,[b,c]]  je dvouprvkový seznam o prvcích a, [b,c] 
[a,b,c,[]]   je čtyřprvkový seznam o prvcích a, b , c  a  [ ] (tj. prázdný seznam) . 

U výkladu  způsobu zápisu seznamů  jsme se zastavili  na tak dlouho proto, že jeho zvládnutí vám ušetří mnoho 
času a problémů. 

6.1. Predikáty prvek a vypust 
Prvním  predikátem pro  práci se  seznamy, který  vytvoříme, bude bude predikát 

   prvek(Prvek,Seznam),  (1) 

který uspěje, je-li Prvek prvkem seznamu Seznam.  Definice predikátu prvek je poměrně snadná 

%  ============================================================== 
%   prvek(X,Sez)         X je prvkem seznamu Sez 
%  ============================================================== 
prvek(X,[X|_]).    %  X je prvkem seznamu, je-li jeho hlavou 
prvek(X,[_|T]):- prvek(X,T).  %  X je prvkem seznamu, je-li prvkem jeho těla 

Díky algoritmu interpretace  prologovských programů se druhá klauzule použije až tehdy,  nepodaří-li se použít 
klauzuli první. Specifikace  predikátu nám  naznačuje,  že  jej lze  použít třemi způsoby: 

-  j a k o  t e s t   (   %  p r v e k ( + A , + B )  )  

     ?- prvek(b,[a,b,c]). 
      yes. 

Nejprve  se   Prolog  neúspěšně  pokusí   použít  první klauzuli,  a tedy  substituovat  X  <- a.  Po neúspěchu 
použije klauzuli  druhou - a hned  úspěšně, neboť prvek b je hlavou těla seznamu [a,b,c]. 

-  p r o  h l e d á n í  p r v ků  s e z n a m u  (  %  p r v e k ( - A , + B )  )  

 ?- prvek(X,[a,b,c]). 
  X=a ; %  pomocí  první  klauzule,  po odmítnutí středníkem se použije  druhá klauzule 
  X=b ;  %  a opět první, která uspěje 
    X=c ;  %  ... atd 
   no 

-  p r o  h l e d á n í  s e z n a m u  (  %  p r v e k ( + A , - B )  )  

 ?- prvek(a,L). 
   L=[a]  ;            %  řešení podle první klauzule 
    L=[_1|[a]]  ;       %  po odmítnutí nové řešení 
    L=[_2,_3|[a]]  ; 
     L=[_4,_5,_6|[a]]    %  atd. 

Při stálém odmítání výsledků středníkem výpočet nikdy neskončí (ve skutečnosti přeteče  paměť). Toto použití 
predikátu prvek nemá samo o sobě význam, může však být, jak ještě uvidíme,  s výhodou  použito k  definování 
složitějších predikátů. 

Jinou velmi potřebnou úlohou je vypouštění prvku ze seznamu. Zkuste si jej naprogramovat ! 

Jistě se vám to povedlo a i  když ne, získali jste lepší předpoklady ke čtení dalšího  textu. Sestrojme  příslušnou 
proceduru společně. 

Vypuštění hlavy je velmi snadné  - získáme tím tělo seznamu. Jinak  hlava  přejde  do  výstupního  seznamu  a  
převedeme úlohu vypouštění z celého seznamu na vypouštění z jeho těla. 

Uvědomme si, že takto pojatá procedura vypouštění uspěje jen tehdy,  když  vypouštěný  objekt   skutečně  je  
prvkem  seznamu. Vyskytuje-li se objekt v seznamu vícekrát, je ze seznamu vypuštěn pouze první výskyt tohoto 
objektu. 

%  ============================================================== 
%  vypust(?X,?S,?L)        Seznam L vznikne vypuštěním prvku X 
%                              ze seznamu S 
%      ?- vypust(b,[a,b,c],L).        L=[a,c] 
%      ?- vypust(X,[a,b,c],L).        X=a  L=[b,c]  ; 
%                                     X=b  L=[a,c]  ; 
%                                     X=c  L=[a,b]  ; 
%                                     no 
%      ?- vypust(d,L,[a,b]).          L=[d,a,b]     ; 
%                                     L=[a,d,b]     ; 
%                                     L=[a,b,d]     ; 
%                                     no 
%  ============================================================== 
vypust(X,[X|T],T).                 %  vypuštěním hlavy seznamu 
                                   %  získáme jeho tělo 
vypust(X,[Y|T],[Y|L]):-            %  nevypouštíme-li hlavu, 
               vypust(X,T,L).      %  vypustíme z těla 

Vzhledem  ke vlastnostem  sestrojeného predikátu  vypust jej lze  použít  jako  alternativu  k  predikátu  prvek,  
můžeme tedy definovat: 

             prvek1(X,Seznam):- vypust(X,Seznam,_). 

Někdy  potřebujeme  vypouštění  poněkud  jiných  vlastností. 

První takovou modifikací je predikát vypust1, který uspěje i v případě, že vypouštěný objekt v seznamu není. 

%  ============================================================== 
%  vypust1(?X,?S,?L)    Seznam L vznikne ze seznamu S případným 
%                         vypuštěním objektu X 
%    ?- vypust1(a,[b,c,d],L).             L=[b,c,d] 
%    ?- vypust1(a,[b,a,c,d],L).           L=[b,c,d] 
%    ?- vypust1(a,[a,b,a,c,d],L).         L=[b,a,c,d] 
%  ============================================================== 
vypust1(X,[X|T],T). 
vypust1(X,[Y|T],[Y|L]):- vypust1(X,T,L). 
vypust1(X,[],[]). 

Druhou modifikací  je predikát vypustvsechny,  který vypustí ze seznamu všechny výskyty daného prvku. 

%  ============================================================== 
%   vypustvsechny(+X,+S,-L)    vypuštěním všech výskytů X ze 
%                              seznamu S vznikne seznam L 
%   ?- vypustvsechny(d,[a,b,c],L).        L=[a,b,c] 
%   ?- vypustvsechny(d,[a,d,b,d,c,d],L).  L=[a,b,c] 
%  ============================================================== 
vypustvsechny(X,[X|T],L):-   %  je-li X hlavou, nepřejde do 
          vypustvsechny(X,T,L). %  výsledku, pokračuji ve 
                                %  vypouštění z těla seznamu 
vypustvsechny(X,[Y|T],[Y|L]):-  %  jestliže se použila tato 
          vypustvsechny(X,T,L). %  klauzule, není X hlavou 
                               %  seznamu v druhém argumentu, 
                                %  stačí vypouštět z těla 
vypustvsechny(X,[],[]).   %  vypuštěním z prázdného seznamu 
    %  získáme opět prázdný seznam 

6.2. Další jednoduché predikáty pro práci se seznamy 
V  tomto odstavci si ukážeme konstrukci několika dalších predikátů pracujících se seznamy. Půjde však spíše 

o programátorská cvičení. Kromě specifikujícího komentáře uvádíme vždy i jedno typické volání daného predikátu. 

a)  p o s l e d n í  p r v e k  s e z n a m u  

%  ============================================================== 
%     posl(+S,-X)   X je posledním prvkem seznamu S 
%      ?-posl([a,b,c,d],X).   X=d. 
%  ============================================================== 
 posl([X],X). 
 posl([X|T],Y):- posl(T,Y). 

%  ============================================================== 
%     poslzbyt(+S,-Zbyt,-X)   X je posledním prvkem seznamu S, 
%                               Zbyt příslušný začátek seznamu 
%    ?- poslzbyt([a,b,c,d],Z,X).   X=d, Z=[a,b,c]. 
%  ============================================================== 
 poslzbyt([X],[],X). 
 poslzbyt([X|T],[X|T1],Y):- poslzbyt(T,T1,Y). 

b)  p r v n í  p r v e k  s e z n a m u  

%  ============================================================== 
%     prv(+S,-X)              X je prvním prvkem seznamu S 
%      ?-prv([a,b,c,d],X).   X=a. 
%  ============================================================== 
 prv([X|_],X). 

%  ============================================================== 
%     prvzbyt(+S,-ZBYT,-X)    X je prvním prvkem seznamu S, 
%                               Zbyt příslušný konec seznamu 
%     prvzbyt([a,b,c,d],Z,X).   X=a, Z=[b,c,d]. 
%  ============================================================== 
 prvzbyt([X|T],T,X). 

c)  p r o s t ř e d n í  p r v e k  s e z n a m u  

%  ============================================================= 
%     prostr(+S,-X)          X je prostředním prvkem seznamu S 
%                              je-li S sudé délky, bere se první 
%                              "ze dvou prostředních prvků" 
%       ?-prostr([a],X).                X = a. 
%       ?-prostr([a,b,c,d,e],X).        X = c. 
%       ?-prostr([a,b,c,d,e,f],X)       X = c. 
%  ============================================================= 

Sestavení příslušného programu není tak snadné. Každého jistě  napadne  řešení  používající  délku  seznamu  a  
spočítání kolikátý prvek  je prostřední. Tuto  ideu však nemůžeme  s našimi dosavadními znalostmi jazyka v  Prologu 
realizovat, protože dosud nemáme k  dispozici vůbec žádné prostředky  pro počítání s čísly. Přesto  však můžeme  úlohu  
vyřešit.  Musíme však  naši proceduru založit na jiném - a elegantnějším - principu: 

V y š l e m e  o d  p o č á t k u  s e z n a m u  s m ě r e m   k  j e h o  k o n c i  d v a  s i g n á l y :  j e d e n   
r y c h l ý  a   d r u h ý  d v a k r á t   p o m a l e j š í .  V   o k a m ž i k u ,  k d y  p r v n í  s i g n á l   d o j d e    
n a   k o n e c   s e z n a m u ,  b u d e   d r u h ý   s i g n á l    p r á v ě  v  p r o s t ř e d k u  s e z n a m u .  

%  ============================================================= 
%     prostr(+S,-X):- prost(+S,+S,-X). 
%               První argument se bude zkracovat dvakrát rychleji 
%                než argument druhý. V okamžiku, kdy první seznam 
%                v prvním argumentu je jeden nebo dva prvky, 
%                bude hlava druhého argumentu prostředním prvkem 
%                původního seznamu 
%       ?-prost([a],[a],X).                        X = a. 
%       ?-prost([a,b],[a,b],X).                    X = a. 
%       ?-prost([a,b,c,d,e],[a,b,c,d,e],X).        X = c. 
%       ?-prost([a,b,c,d,e,f],[a,b,c,d,e,f],X)     X = c. 
%  ============================================================= 
prost([U],[X|T],X).  %  původní seznam byl liché délky 
prost([U,V],[X|T],X).  %  původní seznam byl sudé délky 
prost([U,V|T1],[W|T2],X):-  %  první argument se zkracuje 
 prost(T1,T2,X).  %  dvakrát rychleji než druhý 

d)  r o z d ě l e n í  s e z n a m u  n a  s e z n a m y  l i c h ý c h  a  s u d ý c h  č l e nů  

%  ============================================================== 
%     rozdel(+S,-L1,-L2)   rozděluje seznam S na dva : 
%                            L1 je seznam prvků na lichých místech 
%                              seznamu S 
%                            L2 je seznam prvků na sudých místech 
%                              seznamu S 
%     ?-rozdel([a,b,c,d,e,f,g],S1,S2)   S1=[a,c,e,g] , S2=[b,d,f] 
%  ============================================================== 
rozdel([X,Y|T],[X|T1],[Y|T2]):- rozdel(T,T1,T2). 
Rozdel([X],[X],[]). 
rozdel([],[],[]). 

Predikát rozděl jde používat plnohodnotně „oběma směry“, tedy i pro slévání dvou seznamů. Zkuste si rozmyslet, 
zda to platí i pro jiné z předcházejících predikátů pro práci se seznamy. 

6.3. Zřetězení a obracení seznamu. Použití akumulátoru. 
V  tomto odstavci  sestrojíme několik  predikátů, které mají pro  práci se  seznamy klíčový  význam. Mimoto  se 

budeme zabývat i otázkou výpočetní složitosti jednotlivých predikátů. 

Prvním  z  predikátů  bude  predikát  conc,  k t e r ý  r e a l i z u j e  z ř e t ě z e n í  d v o u  s e z n a mů . 

%  ============================================================== 
%  conc(?L1,?L2,?L)   seznam L vznikne zřetězením seznamů L1 a L2 
%    ?- conc([a,b,c],[1,2,3,4],L).       L=[a,b,c,1,2,3,4] 
%    ?- conc([a,b,c],L,[a,b,c,1,2]).     L=[1,2] 
%    ?- conc(L,[a,b,c],[1,2,a,b,c]).     L=[1,2] 
%  ============================================================== 
conc([],L,L).   %  přiřetězení seznamu L za prázdný seznam dá seznam L 
conc([X|T],L,[X|S]):- %  hlavou zřetězeného seznamu je hlava prvního seznamu, 
  conc(T,L,S). %  jeho tělo vznikne přiřetězením druhého seznamu za 
   %  tělo prvního 

Velmi často potřebujeme seznamy, se kterými pracujeme, o t á č e t . Nejjednoduším způsobem, jak seznamy 
otáčet, je následující predikát obrat, který používá predikát conc: 

%  ============================================================== 
%  obrat(+L1,-L2)    seznam L2 vznikne otočením seznamu L1 
%    ?- obrat([a,b,c],L).           L=[c,b,a] 
%  ============================================================== 
obrat([],[]).  %  Obrácený prázdný seznam je opět prázdný 
obrat([X|T],L):-   %  Obrácený seznam vznikne tak, že 
 obrat(T,T1), %  obrátíme tělo seznamu 
   conc(T1,[X],L).   %  a za něj přiřetězíme jednoprvkový seznam 
    %  obsahující hlavu posledního seznamu 

Zamysleme  se   nad  výpočetní  složitostí   predikátů  conc a obrat. Je zřejmé, že č a s   v ý p o č t u  p r e d i k á t u  
c o n c  r o s t e  l i n e á r n ě  s  d é l k o u  p r v n í h o   s e z n a m u . Pravidlo v  druhé klauzuli se  použije tolikrát, jak 
dlouhý je první seznam. 

 Zkuste si rozmyslet, zda by se zřetězení seznamů dalo realizovat rychleji! 

Protože  predikát obrat  volá predikát  conc tolikrát, kolik činí  délka  obraceného  seznamu,   je  s l o ž i t o s t   
v ý p o č t u   p o d l e  p r e d i k á t u  o b r a t  k v a d r a t i c k á .  

Založíme-li  obracení  seznamu  na  jiném  principu,  než je použití predikátu  conc, můžeme jej realizovat  v 
lineárním čase. Tento  postup je  speciálním případem  tzv. t e c h n i k y  a k u m u l á t o r u , což  je  jeden  z  
nejvděčnějších  obratů  programování v Prologu (i  LISPu).  K  obracení   seznamu  použijeme  ternální  predikát 
concrev: 

%  ============================================================== 
%  concrev(+L,+T,-S)  S je zřetězení obráceného seznamu L 
%                       se seznamem T 
%      tedy predikát ot(+L,-S):- concrev(L,[],S) otáčí seznam 
%      ?-concrev([a,b,c],[1,2],S).    S=[c,b,a,1,2] 
%  ============================================================== 
concrev([],L,L). 
concrev([X|T],L,P):- concrev(T,[X|L],P). 

První argument  predikátu concrev je vstupní,  druhý je také vstupní, hraje však speciální úlohu,  vytváří se v něm 
postupně v jednotlivých  voláních  predikátu  konečná  hodnota,  která se do výstupního parametru zkopíruje, je-li  
splněna koncová podmínka - v tomto případě, že je první vstupní argument již prázdný. 

Protože  je  tento  postup  velmi  důležitý, zkusme trasovat výpočet dotazu 

?- concrev([a,b,c,d],[1,2],L). 

 první argument druhý argument -akumulátor výstupní argument 

 [a,b,c,d]  [1,2]   nekokretizováno 
  [b,c,d]   [a,1,2]  nekokretizováno 
   [c,d]    [b,a,1,2]  nekokretizováno 
  [d]     [c,b,a,1,2]    nekokretizováno 
   []    [d,c,b,a,1,2]  [d,c,b,a,1,2] 

S  použitím  predikátu  concrev  můžeme  pak  definovat nový predikát pro obracení seznamů: 

               obrat1(S,ObrS):- concrev(S,[],ObrS)  , 

který obrací seznam v lineárním čase. 

I  když  se  tím  dopouštíme  velmi  velkého zjednodušování, můžeme vám dát jednu radu: 

M á t e - l i   p r o b l é m y  s   k o n s t r u k c í  n ě j a k é h o   p r o l o g o v s k é h o  p r e d i k á t u ,  
č a s t o  v á m  m ů ž e  p o m o c i   h l e d a t  j i n ý  p r e d i k á t  o  v í c e  p a r a m e t r e c h ,  z e  
k t e r é h o  p o t ř e b n ý  z í s k á t e  v o l á n í m  p o m o c n é h o   p r e d i k á t u   s e   s p e c i á l n í   
i n i c i á l n í  h o d n o t o u  n ě k t e r ý c h  z  n i c h .  

7.Aritmetika 

I  programovací   jazyky,  které  neslouží   k  programování numerických  výpočtů,  se  neobejdou  bez  prostředků  
pro  práci s čísly. Výjimkou  není ani Prolog. Práce  s čísly v něm  však má svoje specifika. Začněme příkladem. 

Jaká bude odpověď na dotaz 

?- X = 1+2.                        ? 

Pokud jste  tipovali, že X=3, pak  jste na  dosavadní výklad nedávali příliš  pozor. Funktor = uspěje,  pokud se oba 
argumenty povede unifikovat. Odpověď tedy bude 

 X = 1+2                             , 

neboť k unifikaci vede substituce X <- 1+2 . 

To moc nepotěší.  Pokud chceme, aby se provedl  výpočet a ne unifikace, musíme použít speciálního operátoru is. 
Tedy 

?- X is 1+2. 
 X = 3 

Cíl s použitím operátoru is může mít tvar: 

  X is V , 

kde  X je  proměnná a  V je  aritmetický výraz. V š e c h n y  p r o m ě n n é , které se  (případně) vyskytují v e   v ý r a z u  
V ,  m u s í   b ý t  v  o k a m ž i k u  s p l ňo v á n í  c í l e  k o n k r e t i z o v á n y  n a   č í s e l n o u  h o d n o t u , jinak dojde 
k běhové chybě. Obdobně  dojde k běhové chybě, pokud  proměnná X je již konkretizována, avšak na nečíselnou 
hodnotu. 

Při splňování  daného cíle se  vyčíslí hodnota aritmetického výrazu V - nechť je to z - a potom 

-  p o k u d    p r o m ě n n á   X    n e n í   k o n k r e t i z o v á n a , 
 provede  se substituce  X < z  a cíl uspěje 

-  p o k u d  p r o m ě n n á   X  j e   k o n k r e t i z o v á n a  n a   č í s e l n o u  h o d n o t u   (označme ji y), 
 porovnají se  hodnoty z a y. Pokud jsou stejné, cíl uspěje, jinak selže. 

 Aritmetické  vyhodnocení  vynucují   i  aritmetické  relační operátory: 

          =:=            aritmetická rovnost, 
          =\=            aritmetická nerovnost, 
          <               je menší, 
          >               je větší, 
          =<             je menší nebo rovno, 
          >=             je větší nebo rovno. 

Na   obou  stranách   od  relačního   operátoru  musí   stát aritmetický  výraz,  jehož  všechny   proměnné  jsou  v  
okamžiku splňování  cíle již  konkretizovány na  číselnou hodnotu. Není-li tomu tak, dojde k běhové chybě. 

Při  pokusu o  splnění cíle  konstruovaného z  aritmetických relačních  operátorů  se  vyhodnotí   oba  výrazy  a  
provede  se příslušné porovnání získaných číselných  hodnot - jsou-li v daném vztahu, cíl uspěje, nejsou-li, selže. 

Ilustrujme právě vyloženou látku na krátké konverzaci: 

?- X is Y+1.  %  běhová chyba - nedefinovaný aritmetický výraz 

?- 2+3 = 3+2. 
     no      %  tyto dva termy nelze unifikovat 

?- 2+3 =:= 3+2. 
     yes.            %  oba aritmetické výrazy mají stejnou hodnotu 

?- X is 2+3, Y is 2*X , Z is 12-3, Y is Z . 
     no            %  za jednotlivé proměnné se postupně substituovala čísla   
   %  X=5 , Y=10  a Z = 9,  při vyhodnocování predikátu Y is Z   
   %  byla tedy proměnná jY již konkretizována  a její hodnota   
   %  různá od hodnoty proměnné Z 

?- X<5.   %  běhová chyba - výraz vlevo nelze vyčíslit 

?- X is 5 , X is X+1 . 
     no  %  za X substituovalo 5, což není rovno 6 

Viděli  jsme  tedy,  ž e   o p e r á t o r   i s   n e n í  p ř e s n o u  a n a l o g i í  " p ř i ř a z o v á t k a "  : =   , které 
znáte  z Pascalu. Jestliže  již jednou byla  proměnná  X  konkretizována,  není  již  možno později tuto substituci  změnit 
(při  postupu výpočtu  vpřed). K  takové změně může  dojít  jen  tak,  že  je  při  navracení  výpočtu zpět tato substituce 
odvolána. 

Z významu standardního operátoru is a relačních operátorů 

 ( = : =  ,  = \ =  ,  <  ,  > ,  =<   ,  >=  )  

je zřejmé, že nemá dobrý význam, aby  se  je  pokoušel  interpret  splňovat  znovu hledáním jiných hodnot příslušných 
aritmetických výrazu.  V tomto smyslu mluvíme o tom, že aritmetické operátory nebacktrackují. 

Vytvořme  s  pomocí  aritmetických  operátorů  několik velmi jednoduchých predikátů. 

%  ============================================================== 
%   soucet(+Sezn,?Souc)   uspěje, je-li Sezn seznam utvořený 
%                           z čísel a spočítá  do proměnné Souc 
%                           součet prvků tohoto seznamu 
%  ============================================================== 
soucet([],0).  %  součtem prázdného seznamu je nula 
soucet([N|L],S) :-  %  součet seznamu vznikne přičtením 
 soucet(L,S1), S is S1+N. %  součtu jeho těla k hlavě 

Cvičení. Co "budou počítat" následující modifikace predikátu soucet ? 

  soucet1([],0). 
  soucet1([N|L],S) :-  S = S1 + N , soucet1(L,S1). 

  soucet2([],0). 
  soucet2([N|L],S) :-  S is S1 + N , soucet2(L,S1), . 

  soucet3([],0). 
  soucet3([N|L],S) :-  soucet3(L,S1), S = S1 + N . 

V  prologu n e m á m e   k   d i s p o z i c i   datovou strukturu  p o l e , na kterou jsme  zvyklí z jiných  
programovacích jazyků. Nahradit  je mohou  seznamy, musíme  si však  uvědomit, že  nemáme k dispozici mechanismus 
indexování a přístup k jednotlivým prvků seznamu trvá rozdílně dlouho. P ř í s t u p  k  n - t é m u  p r v k u  seznamu 
zprostředkuje následující procedura. 

%  ============================================================== 
%   nty(+N,?Sezn,?K)      K je N-tým prvkem seznamu Sezn 
%     ?- nty(3,[a,b,c,d],X).        X=c 
%  ============================================================== 
nty(1,[H|_],H). 
nty(N,[_|L],K):- N1 is N-1, nty(N1,L,K). 

Cvičení 

   Sestavte  další procedury  umožňující provádět  na seznamech operace, které používáme u polí: 

dosad_za_nty(+N,?K,?Sezn,?NSezn) 
  %  seznam NSezn vznikne ze seznamu Sezn tak, že změníme N-tý prvek  

  %  na hodnotu K 

vypust_nty(+N,?Sezn,?NSezn) 
 %  seznam NSezn vznikne ze seznamu Sezn vypuštěním N-tého prvku 

vloz_nty(+N,?K,?Sezn,?Nsezn) 
  %  seznam NSezn vznikne ze seznamu Sezn vložením prvku K na N-té místo 

Často  potřebujeme zjistit  d é l k u  s e z n a m u ,  to zprostředkuje následující procedura: 

%  ============================================================== 
%   delka(+Sezn,?Del)     uspěje, je-li Sezn seznam, Del je 
%                           počet jeho prvků 
%     ?-delka([a,b,c],N).           N = 3 
%  ============================================================== 
delka([],0). 
delka([H|L],N):- delka(L,N1), N is N1+1. 

Jistě  nejpopulárnějším  číselným  algoritmem  je  Eukleidův algori tmus, zjišťující nejmenšího společného 
dělitele dvou čísel. Uvádíme dvě verze (z pedagogických důvodů bez komentářů - doplňte si je !). 

nsd(A,A,A). 
nsd(A,B,D):- A>B, A1 is A-B, nsd(A1,B,D). 
nsd(A,B,D):- A<B, B1 is B-A, nsd(A,B1,D). 

nsd1(A,A,A). 
nsd1(A,B,D):- A>B, A1 is A-B, nsd1(A1,B,D). 
nsd1(A,B,D):- nsd(B,A,D). 

Cvičení 

    Sestavte predikát   nsdz(+Sezn,?N)  , který pro seznam čísel Sezn spočítá největšího společného dělitele  
    jeho prvků. 

Základem  algoritmů vnějšího  třídění je  algoritmus s l é v á n í  d v o u  u s p o ř á d a n ý c h  p o s l o u p n o s t í  d o  
j e d n é . Naprogramujme ho: 

%  ============================================================== 
%   slej(+S1,+S2,-S)       uspořádaný seznam S vznikne slitím 
%                          uspořádaných seznamů S1 a S2 
%     ?-slej([2,4,8],[1,2,3,5],L).   L=[1,2,2,3,4,5,8] 
%  ============================================================== 
slej([X|T],[Y|S],[X|L]):- X<Y, slej(T,[Y|S],L). 
slej([X|T],[Y|S],[Y|L]):- Y<X, slej([X|T],S,L). 
slej([X|T],[X|S],[X,X|L]):- slej(T,S,L). 
Slej([X|T],[],[X|T]). 
Slej([],[X|T],[X|T]). 
slej([],[],[]). 

Procedura  slej  má  jednu  zajímavou  vlastnost  - v každém případě lze použít právě  jednu  z jejích klauzulí. 
To velmi pomáhá zejména  při ladění  programu. Takovým  procedurám se  někdy říká d e t e r m i n i s t i c k é .  
Pokud bychom  o tuto  vlastnost nestáli,  mohli bychom poslední tři klauzule nahradit následujícími dvěma: 

slej(T,[],T). 
slej([],T,T). 

7.1. Třídící algoritmus quicksort 

Na závěr této kapitoly si  ukážeme, jak snadno se v Prologu napíše algoritmus  třídění quicksort. 

Quicksort je rekursivní algoritmus,  který na základě čísla, kterému  se obvykle  říká p i v o t ,  rozdělí tříděný  
seznam na  dvě části. V  jedné jsou čísla  menší než pivot,  v druhé zbylá.  Obě tyto  části  pak  setřídí  stejným  
postupem  (část  délky  jedna pochopitelně již třídit netřeba). 

Rozdělení   seznamu  podle   pivota  realizuje   následující predikát: 

%  ============================================================== 
%   split(+Pivot,+Vstup,-Male,-Velke)   rozdělí vstupní seznam 
%                       Vstup na dva seznamy : 
%                Male  je tvořen prvky menšími než Pivot 
%                Velke je tvořen prvky většími nebo rovnými než Pivot 
%      ?- split(6,[12,4,2,87,10,3],U1,U2). 
%                         U1=[4,2,3]   U2=[12,87,10]] 
%  ============================================================== 
split(_,[],[],[]).  %   rozdělením prázdného seznamu 
   %   dostaneme dva prázdné seznamy 
split(Pivot,[Y|Zbytek],[Y|Male],Velke) :- 
               Pivot>Y,    %  Y patří mezi malé 
               split(Pivot,Zbytek,Male,Velke). %   rozdělení zbytku 
split(Pivot,[Y|Zbytek],Male,[Y|Velke]) :- 
               Pivot=<Y,  %  Y patří mezi velké 
               split(Pivot,Zbytek,Male,Velke). %   rozdělení zbytku 

S použitím obvyklého predikátu pro zřetězení seznamů conc již snadno zkonstruujeme  třídící predikát quicksort  
(za pivota jsme vzali první prvek seznamu). 

%  ============================================================== 
%   quicksort(+Vstup,-Setrideno) 
%                         Setrideno je setříděný seznam Vstup 
%  ============================================================== 
quicksort([],[]). 
quicksort([Pivot|Zbytek],Setrideno) :- 
           split(Pivot,Zbytek,Mala,Velka), 
           quicksort(Mala,SetridenaMala), 
           quicksort(Velka,SetridenaVelka), 
           conc(SetridenaMala,[Pivot|SetridenaVelka],Setrideno). 
Conc([],L,L). 
conc([X|L1],L2,[X|L3]) :- 
          conc(L1,L2,L3). 

K tomuto  programu se ještě  vrátíme v deváté  kapitole, při diskusi o efektivitě prologovských programů. 

8. Eratosthenovo síto 

V našem výkladu Prologu  jsme již pokročili natolik, abychom mohli  sestavit  program  řešící   reálnou  úlohu  
z aritmetiky. Vybereme  si  implementaci  jednoho  z  nejstarších  a nejhezčích postupů, jak zjišťovat prvočísla  - tzv. 

algoritmu Eratosthenova síta .   Tento  algoritmus  hledá  prvočísla  tak,  ž e   z e   s e z n a m u  k a n d i d á t ů  
s y s t e m a t i c k y   e l i m i n u j e  č í s l a ,  k t e r á   p r v o č í s l y  n e j s o u . Půvab  algoritmu spočívá  především v  tom, 
že  k jeho  provádění n e m u s í m e  vůb e c  u m ě t  d ě l i t . 

P ř i p o m eňm e  s i  p o s t u p  E r a t o s t h e n o v a  s í t a : 

Máme dáno  číslo N a chceme  zjistit všechna prvočísla menší nebo  rovna  než  N.  Stačí  se  j i s t ě   o m e z i t  
n a  h l e d á n í  l i c h ý c h  p r v o č í s e l , jediným sudým prvočíslem je číslo 2. 

Vyjdeme ze seznamu kandidátů,  který budou na počátku tvořit všechna lichá čísla menší než   N. O d t r h n e m e  
z e  s e z n a m u  p r v n í  p r v e k  ( n y n í  j e  t o  3 ) ,  k t e r ý  j e  j i s t ě  p r v o č í s l e m , a upravíme seznam tak, že z něj  
v y š k r t á m e  v š e c h n y  n á s o b k y  t o h o t o  p r v n í h o   p r v k u  (k tomu stačí umět  jen přičítat). Po  této úpravě 
dostáváme  n o v ý  s e z n a m  k a n d i d á t ů .  J e h o   p r v n í   p r v e k   j e   o p ě t   p r v o č í s l e m . C e l ý  p o s t u p  
b u d e m e    o p a k o v a t ,  dokud   neprojdeme  celý   seznam  kandidátů. (Přesvědčte se  sami, že navržený algoritmus  
skutečně řeší danou úlohu). 

Navržený  v ý p o č e t  mů ž e m e   u r y c h l i t , pokud  si uvědomíme,  že vyškrtávání můžeme ukončit v 
okamžiku,  kdy první číslo v seznamu kandidátů  je větší  než druhá  odmocnina z  N.  

Má-li totiž číslo K (  <=N )  jednoho dělitele  většího než  N , musí  mít i jiného dělitele menšího než  N . 

Následující program realizující Eratosthenovo síto je velmi podrobně  komentován, přesto  si však  zaslouží 
několik poznámek, protože jsme v něm p o u ž i l i  n ě k o l i k  t e c h n i c k ý c h  " f í g l ů " .  

Hlavním  predikátem  je  nulární  predikát  prvoc,  který se nejprve uživatele zeptá  na číslo N, po té  vytvoří 
pomocí volání predikátu  seznam seznam  kandidátů na  prvočísla, po  té provede vlastní algoritmus  Eratosthenova síta 
a  naonec zavolá predikát vypis, který  způsobí  výstup  požadovaného  seznamu  prvočísel  (po  10 prvočíslech na 
řádce). 

Vlastní  algoritmus Eratosthenova  síta realizuje  predikát udelej, který má tři argumenty 

udelej(N,S,List):- vytvor(N,S,[],List) 

S je seznam kandidátů, List je  hledaný seznam (ovšem v obráceném tj.  klesajícím  pořadí).  Důvodem   pro  takovou  
konstrukci  je skutečnost, že v Eratosthenově sítu zjišťujeme prvočísla v pořadí od nejmenšímu k největšímu. 
Kdybychom chtěli mít seznam prvočísel stále  rostoucí, museli  bychom nové  prvočíslo přidávat  vždy na konec  
aktuálního  seznamu,  což  je  v  Prologu zbytečně pomalé. Přidáváme tedy raději na začátek  seznamu prvočísel - a 
dostáváme tak  seznam  v  opačném  pořadí.  Ten  pak  na  závěr (algoritmem s lineární složitostí) otočíme. 

Jak  vidíme, provádí  celou práci  predikátu udelej predikát vytvor. Jeho třetí argument je použit jako akumulátor, 
ze kterého se   hodnota   zkopíruje   do   čtvrtého  (výstupního)  argumentu v okamžiku,  kdy  se  druhý  argument  
(seznam  kandidátů)  stane prázdným. K  tomu však většinou  nedojde (dojde k  tomu vůbec ?), protože  pokud se  zjistí, 

že  první prvek  seznamu kandidátů  je větší než  N , pak již víme,  že všechna čísla v seznamu kandidátů jsou 
prvočísly  a můžeme jej rovnou  přidat k hodnotě akumulátoru (pomocí predikátu concrev, který provede i nutné 
otočení). 

Pokud je  první kandidát ještě  menší než  N,  pak se pomocí predikátu  vyh  provede  vyškrtání  násobků  tohoto  
kandidáta ze seznamu  kandidátů.  Predikát  vyh  dostává  v  prvním  argumentu hodnotu  prvního kandidáta,  v druhém  
číslo od  kterého se má ve třetím argumentu  vyškrtávat. Druhý argument  je inicializován na dvojnásobek prvního 
kandidáta. 

Cvičení 

Predikát  vyh je  v programu  zkonstruován tak,  že se snaží vypouštět i sudé násobky prvního  kandidáta. 
      To je však zbytečné, neboť v  seznamu kandidátů nejsou  žádná sudá čísla.  Modifikujte program tak, aby  
      vyškrtával jen liché násobky. 

Pr o g r a m E r a t o s t h e n o v o  s í t o  
%  ============================================================== 
%  prvoc    hlavní   procedura,  která   obsluhuje  i   komunikaci 
%           s uživatelem - vyžádá si vstup přirozeného čísla N - 
%           a pak spočítá a vytiskne všechna prvočísla menší nebo 
%           rovná tomuto číslu 
%  ============================================================== 
prvoc:- write($Zadej cislo:$), 
      read(N), N>1, 
      seznam(N,S), %  S je seznam lichých čísel menších nebo rovných než N 
      udelej(N,S,ObrPrvoc), %  vlastní Eukleidův algoritmus 
      concrev(ObrPrvoc,[ ],Prvoc), %  seznam musíme otočit 
      vypis(10,[2|Prvoc]).  %  výstup zjistěných prvočísel 

%  ============================================================== 
%  seznam(+N,-S)  vygeneruje do S seznam lichých čísel od 3 do N 
%  ============================================================== 
seznam(N,S):- sezn(3,N,S). 

%  ============================================================== 
%  sezn(+K,+N,-T)  T seznam čísel K, K+2, K+4, ... do N 
%  ============================================================== 
sezn(N,N,[N]).   %  ukončení seznamu 
sezn(K,N,[ ])   :- K>N.  %  K již do seznamu nepřijde 
sezn(K,N,[K|L]):- K<N,  %  K bude hlavou 
                  K1 is K+2,  %  další prvek bude o 2 větší 
                  sezn(K1,N,L). 

%  ============================================================== 
%  vypis(+Max,+List)    vypisuje prvky seznamu List, na každé 
%                       řádce bude Max prvků 
%  ============================================================== 
vypis(Max,List):- vypis(Max,0,List). 

%  ============================================================== 
%  vypis(+Max,+K,+List) vypisuje prvky seznamu List, na každé 
%                       řádce bude Max prvků, K je počet prvků, 
%                       které již vystoupili na aktuální řádce 
%  ============================================================== 
vypis(_,_,[ ]):-nl. %  výpis prázdného seznamu na daném 
vypis(Max,Max,L):- nl, %  řádku již vystoupilo Max čísel 
            vypis(Max,0,L). %  výstup na další řádek 
vypis(Max,K,[H|L]):- K<Max, 
            write(H), write(' '), %  výstup čísla H 
            K1 is K+1, %  zvětšení počítadla 
            vypis(Max,K1,L). 

%  ============================================================== 
%  udelej(+N,+S,-List)  pokud je S rostoucí seznam všech lichých 
%          čísel menších nebo rovných N (tj. kandidátů na prvo- 
%          čísla), bude List klesající seznam všech prvočísel 
%          menších než N 
%  ============================================================== 
udelej(N,S,List):- vytvor(N,S,[ ],List). 

%  ============================================================= 
%  vytvor(+N,+S,+Acum,-List)  realizace predikátu udelej pomocí 
%                             akumulátoru ve třetím argumentu 
%                             druhý argument vstup, čtvrtý výstup 
%  ============================================================== 
vytvor(N,[ ],L,L).  %  je-li druhý argument prázdný seznam, 
   %  zkopíruje se třetí argument (akumulátor) do čtvrtého argumentu 
vytvor(N,[H|T],L,P):- H*H=<N, %  H je dost malé 
  H2 is H+H,  %  H2 je dvojnásobek H 
   vyh(H,H2,T,T1),  %  T1 vznikne vyhozením násobků  čísla H z T 
    !,   %  zaříznutí pro jistotu, viz kap. 11 
   vytvor(N,T1,[H|L],P). %  vstupem ve druhém  argumentu se stal   
     %  vyškrtaný seznam kandidátů T1, 
     %  H bylo přidáno jako hlava do akumulátoru 
vytvor(N,[H|T],L,P):- H*H>N,  %  hlava H vstupního seznamu je již příliš velká 
  concrev(T,[H|L],P).  %  před nashromážděný seznam  [H|L]  v   
     %  akumulátoru je nutné  přidat otočený zbytek 
      %  vstupního seznamu T 

%  ============================================================== 
%   vyh(+H,+P,+T,-T1)  Seznam T1 vzniká z rostoucího seznamu T 
%                        vypuštění všech čísel tvaru 
%                        P+k*H pro k=0,1,..... 
%  ============================================================== 
vyh(_,_,[],[]).  %  vyhozením z [ ] vznikne [ ] 
vyh(H,Co,[Co|T],L):- Co1 is Co+H,  %  vyhodil jsem hlavu Co 
                      vyh(H,Co1,T,L).  %  budu vyhazovat prvek Co1 
vyh(H,Co,[X|T],L):-  Co<X, Co1 is Co+H, %  Co není v seznamu 
                      vyh(H,Co1,[X|T],L). %  budu vyhazovat Co1 
vyh(H,Co,[X|T],[X|L]):- Co>X, %  hlava X zůstává 
                       vyh(H,Co,T,L). %  v seznamu 

%  ============================================================== 
%  concrev(+L,+T,-S)  S je zřetězení obráceného seznamu L 
%                       se seznamem T 
%  ============================================================== 
concrev([],L,L). 
concrev([X|T],L,P):- concrev(T,[X|L],P). 

9. Efektivita prologovských programů 

     V   této  kapitole   se  budeme   věnovat  několika  ukázkám efektivních implementací algoritmů v  Prologu. K 
této tématice se ještě vrátíme v kapitole 11 pojednávající o operátoru řezu. 

9.1. Fibonacciova posloupnost 

S tzv. Fibonacciovou  posloupnost í ,  

   1 , 1 , 2 , 3 , 5 , 8 , 13 , 21 , 34 , 55 , 89 , 144 , ... 

definovanou rekursivními vztahy 

 f(1) = 1 
 f(2) = 1 
 f(n+2) = f(n) + f(n+1)        , 

se  můžete  setkat  velmi  často  nejen  při  analýze  složitosti algoritmů, ale i v mnoha jiných překvapivých souvislostech 
(např. tzv.  zlatý  řez  -  "dokonalý"  poměr  stran  obrazu,  tak ctěný renesančními malíři - jde odvodit  z této 
posloupnosti). O to nám však  nejde.  Program,  který  vznikne  doslovným  přepisem  této definice,  má 
exponenciální   s loži tost  a  může tedy  sloužit jako školní  příklad,  jak  se  rekurse  užívat  nemá. V Prologu tento 
doslovný přepis vypadá takto: 

 fib(1,1).  fib(2,1). 
     Fib(N,F) :- N>2, 
   N1 is N-1, fib(N1,F1), 
          N2 is N-2, fib(N2,F2), 
          F is F1+F2. 

V  učebnicích programování  najdete závěr,  že v  takovýchto případech  je  lepší  používat  místo  rekurse  iteraci.  
Co však v Prologu, který je  na rekursi "bytostně" založen ?  I v něm jde pochopitelně    efektivní    (lineární)    výpočet   
Fibonacciovy posloupnosti naprogramovat. 

% ================================================================ 
%  fibiter(+K,+N,+Predch,+Aktual,-Hodnota) 
%           počítá Fibonacciovu posloupnost iterační metodou 
%           předpokládá,že v argumentu Aktual je K-tý a v argumentu 
%           Predch (k-1)-ní člen Fibonacciovy posloupnosti 
%           jsou-li tyto předpoklady při volání splněny, 
%           spočte do výstupního parametru Hodnota N-tý člen 
%           Fibonacciovy posloupnosti,můžeme tedy definovat 
%   fib3(N,F):- fibiter(2,N,1,1,F). 
% ================================================================ 
fibiter(N,N,_,F2,F2).  %  Pokud je čítač v prvním argumentu roven N,  pak již je   
    %  hodnota  N-tého členu ve čtvrtém argumentu 
fibiter(M,N,F1,F2,F):- M<N,  %  Pokud je čítač ještě menší než N, 
         NextM is M+1,  %   pak třeba zvětšit čítač o 1 
         NextF2 is F1+F2,  %  a spočíst novou hodnotu 
         fibiter(NextM,N,F2,NextF2,F). %  a celý proces opakovat 
     %  se zvětšeným čítačem 
fib3(N,F):-   %  výpočet Fibonacciovy posloupnosti 
   fibiter(2,N,1,1,F).  %  se převede na volání iteračního predikátu fibiter  
   %  s příslušnou  inicializací vstupních parametrů 

Cvičení 

Naprogramujte iterativní  výpočet Fibonacciovy posloupnosti, ve  kterém   se  v  každém  kroku   spočítají  
    dvě  nové  hodnoty posloupnosti. 

9.2. Zřetězení seznamů a rozdílové seznamy 

Tento odstavec je poněkud obtížnější, proto vám doporučujeme vrátit se k němu později ještě jednou. 

Podařilo  se vám  realizovat zřetězení  seznamů rychleji než lineárně ? Asi ne, ale přesto  to jde. Musíme ovšem 
přejít k jiné reprezentaci   seznamů,   k   r o z d í l o v ý m    s e z n a mům ,  které  jsou speciálním případem n e ú p l n ě  
d e f i n o v a n ý c h  d a t o v ý c h  s t r u k t u r .  N e ú p l n á  d a t o v á  s t r u k t u r a  j e  t e r m ,  k t e r ý  o b s a h u j e  
n ě j a k o u  “ v o l n o u ”  p r o m ě n n o u  t j .  p r o m ě n n o u ,  z a  k t e r o u  j e š t ě  n e b y l o  s u b s t i t u o v á n o .  

Budeme seznamy  reprezentovat jako "rozdíly", kde menšenec je seznam, ve kterém nahradíme atom [] 
proměnnou, a menšitel je tatáž proměnná. Seznam [a,b,c] bude tedy reprezentován termem 
                           [a,b,c|L]-L  . 

K e  z ř e t ě z e n í  r o z d í l o v ý c h  s e z n a mů  s tačí překvapivě  jediný  fakt.  Zřetězení  tedy l z e  
r e a l i z o v a t  v  k o n s t a n t n í m  č a s e . Stačí k tomu fakt: 

  conc1(A-B,B-C,A-C). 

Ukažme si, jak probíhá zřetězení seznamů  [ a, b, c |  T1 ] - T1   a    [ e, f  |  T2 ]  -  T2   pomocí predikátu conc1. Dojde 
k substitucím: 

    A <-  [ a, b, c |  T1 ]  ,  
  B <- T1  ,  B <-  [ e, f  |  T2 ]  , 

  tedy musí platit   T1 <-  [ e, f  |  T2 ]    a    A <-  [ a, b, c |  [ e, f  | T2 ] ] ,  což není nic jiného než   
  A <-  [ a, b, c,  e, f  | T2 ]   . 

Výsledkem tedy bude [ a, b, c,  e, f  | T2 ]   -  T2  . 

Uvědomte si, že pokud za volnou proměnnou v rozdílovém seznamu substituujeme seznam, např. 
 [a,b,c|[d,e]]-[d,e] , 

není daný term již rozdílovým seznamem (neobsahuje volnou proměnou) .   

Sestavme ještě proceduru pro převod seznamu na rozdílový a zpět.   

%  ============================================================== 
%   prevod(?ObycSezn,?RozdSezn)  
%  převádí seznam ObycSezn na reprezentaci pomocí rozdílových 
%     seznamů RozdSezn T1 a zpět  
%  ?-prevod([a,b,c],RS). RS=[a,b,c|T]-T  
%  ?-prevod(S,[a,b,c|T]-T). S=[a,b,c] 
%  ============================================================== 
prevod([],T-T):- var(T). % predikat var(T) uspěje, pokud T je   
   % dosud volná proměnná viz. kapitola 16   
prevod([X|L],[X|T1]-T):- var(T),prevod(L,T1-T). 

Pro převod ze seznamu na rozdílový bychom mohli volání predikátu var vynechat. 

V   příštím   paragrafu    použijeme   rozdílových   seznamů k odstranění použití predikátu conc z procedury 
quicksort. 

9.3. Efektivnější implementace quicksortu 
V  odstavci  7.1.  jsme  sestrojili  prologovskou proceduru, která realizuje  algoritmus quicksortu. P o u ž i t í   

p r e d i k á t u  c o n c  v této  p r o c e d u ř e   z p ů s o b u j e    j e j í   z b y t e č n o u   s l o ž i t o s t .  Použití predikátu  conc  
je  však   v  tomto  algoritmu  zbytečné,  pokud použijeme neúplně definované datové struktury. 

%  ============================================================== 
%            Quicksort bez použití predikátu conc 
%  ============================================================== 
%  quick(+Vstup,-Usporadany,+Pomocny) 
%       Relace "quick" má tento význam: 
%           quick(A,B,C)=sort(A,P),conc(P,C,B) 
%               tedy seznam B vznikne připojením seznamu C za 
%               seznam P, který  vznikl uspořádáním seznamu A. 
%    ?-quick([4,2,7,1],B,[5,6,9]).   B=[1,2,4,7,5,6,9] 
%  Definujeme-li tedy predikát 
%               quicksort1(A,B):-quick(A,B,[]). 
%  dostáváme třídící algoritmus 
%  ============================================================== 
quick([Pivot|Zbytek],S,Konec):- 
  split(Pivot,Zbytek,Mala,Velka),  % Predikát split je stejný jako v odstavci 7.1 
  quick(Mala,S,[Pivot|Y]), 
  quick(Velka,Y,Konec). 
quick([],Konec,Konec). 

quicksort1(Vstup,Setrideno) :- 
  quick(Vstup,Setrideno,[]).. 

Stejnou ideu můžeme popsat i v řeč i  rozdí lových seznamů : 

%  ============================================================== 
%    Quicksort bez použití predikátu conc s rozdílovými seznamy 
%  ============================================================== 
quick2([Pivot|Zbytek],A1-Z2) :- 
               split(Pivot,Zbytek,Mala,Velka), 
               quick2(Mala,A1-[Pivot|A2]), 
               quick2(Velka,A2-Z2). 
quick2([],Z-Z). 

10. Stromy 

Rekursivní  charakter   jazyka  a  to,   že  " p o č í t á   p ř í m o  s  t e r m y " ,   dělá   z  Prologu  velmi  vhodný   
nástroj  pro  práci s datovými   strukturami,  které   jsou  definovány   rekursivně. Nejdůležitějším případem  takových 
struktur jsou  binární  stromy . Můžeme je reprezentovat například takto: 

B i n á r n í m  s t r o m e m  rozumíme 
 buď    a tom    nil   reprezentující prázdný strom 
 nebo strukturu  t(L,V,R) , 
       kde V je vrchol stromu, 

     L je levý podstrom a 

   R je pravý podstrom , 
 reprezentující binární strom 

Následujícímu stromu 

 
   a 
 

  b  c 
 

   e  f g 
 
    h 

odpovídá prologovský term 

t( t(nil,b,t(nil,e,nil)),  %  levý podstrom 
   a,    %  vrchol 
   t(t(nil,f,t(nil,h,nil)),c,t(nil,g,nil)) ) %  pravý podstrom . 

Abychom ho  nemuseli opakovaně zadávat  z klávesnice, můžeme si ho "uložit do programu" klauzulí 

nas_strom( t( t(nil,b,t(nil,e,nil)), 
           a, 
           t(t(nil,f,t(nil,h,nil)),c,t(nil,g,nil)) )  ). 

10.1. Průchody stromem 

     Velmi  často  potřebujeme   " p r o j í t "   s t r o m e m ,  abychom  buď provedli  pro každý  prvek obsažený  ve 
stromě  nějakou akci nebo abychom  převedli prvky  stromu v  nějakém specifickém  pořadí do seznamu či na výstup. 

Nejprve sestrojíme jednu proceduru,  která  p r o c h á z í  s t r o m  d o  h l o u b k y .  

%  ============================================================== 
%    preorder(+Strom,-Seznam)   průchod stromem do hloubky 
%  ============================================================== 
preorder(t(L,V,R),[V|S]):- 
     preorder(L,SL), 
     preorder(R,RL), 
     conc(SL,RL,S). 
preorder(nil,[]). 

Na dotaz 

?- nas_strom(T),preorder(T,L). 

pak dostaneme odpověď 

T = ....... 
L = [a,b,e,c,f,h,g]      , 

tedy skutečně   p r ů c h o d  s t r o m u  d o  h l o u b k y . 

Na první pohled se zdá, že naprogramování procházení stromem do  šířky  musí  být  v  Prologu  podstatně  
obtížnější než průchod do hloubky, který máme v Prologu díky "vestavěnému" backtrackingu "zdarma". To je sice 
pravda, ale zvolíme-li  vhodný úhel  pohledu, l i š í   s e  a l g o r i t m y   p rů c h o d u  d o  h l o u b k y  a  d o  š í řk y  
j e n  p o u ž i t o u  p o m o c n o u  d a t o v o u  s t r u k t u r o u .  

Pro  průchod budeme používat p o m o c n ý   s e z n a m , ve kterém si budeme  pamatovat  " r o z p r a c o v a n é   
p o d s t r o m y " .  

�� Začneme tím, že do tohoto seznamu dáme celý vstupní strom. 
�� Jeden  krok  algoritmu pak  spočívá  v  tom, z  pomocného seznamu   vyzvedneme  první  prvek  -  to  je  

nějaký  strom  t(L,V,R)  - a  pošleme  V  na  výstup  a  podstromy  L  a R přidáme do  pomocného seznamu. 
�� Algoritmus končí vyčerpáním pomocného seznamu. 

%  ============================================================== 
%   projdi(+Strom,-Vystup)     projde stromem Strom a vydá jeho 
%                               prvky do seznamu Vystup 
%     Pořadí prvků v seznamu výstup závisí na použité  
%          proceduře pridej 
%  ============================================================== 
projdi(nil,[]).  %  projít prázdný strom 
projdi(Strom,Vysl):-   %  projít neprázdný strom 
     zpr([Strom],Vysl).  %  inicializace pomocného seznamu 
zpr([],[]).   %  ukončující klauzule 
zpr([t(L,V,R)|Z],[V|Kon]):-  %  vrchol V do výstupu 
     pridej(L,R,Z,Z1),zpr(Z1,Kon). %  podstromy L a R do pomocného seznamu 

Přidáváme-li podstromy  na začátek pomocného  seznamu, chová se tento jako z á s o b n í k , a dostáváme opět 
procházení do hloubky. 

%  =============================================================== 
%     pridej(+Levy,+Pravy,+Kam,-Vysl) přidá podstromy Levy a Pravy 
%                    na začátek seznamu Kam, vznikne seznam Vysl 
%  =============================================================== 
pridej(L,R,Z,Z1):-pridej_na_zac(L,R,Z,Z1). 
pridej_na_zac(nil,nil,Z,Z).   %  prázdné stromy do seznamu nedávám 
pridej_na_zac(nil,R,Z,[R|Z]). 
pridej_na_zac(L,nil,Z,[L|Z]). 
pridej_na_zac(L,R,Z,[L,R|Z]). 

?- nas_strom(T),projdi(T,L).  %   po dotazu  dostaneme 
T = ....... 
L = [a,b,e,c,f,h,g]  %  průchod do hloubky 

Přidáváme-li podstromy na konec  pomocného seznamu, chová se tento jako f r o n t a , a dostáváme procházení stromu 
do šířky. 

%  =============================================================== 
%     pridej(+Levy,+Pravy,+Kam,-Vysl) přidá podstromy Levy a Pravy 
%                    na konec seznamu Kam, vznikne seznam Vysl 
%  =============================================================== 
pridej(L,R,Z,Z1):-pridej_na_kon(L,R,Z,Z1). 
Pridej_na_kon(nil,nil,Z,Z). 
pridej_na_kon(nil,R,Z,Z1):- conc(Z,[R],Z1). 
pridej_na_kon(L,nil,Z,Z1):- conc(Z,[L],Z1). 
pridej_na_kon(L,R,Z,Z1):- conc(Z,[L,R],Z1).  . 

?- nas_strom(T),projdi(T,L).         %   po dotazu 
                                     %   dostaneme 
T = ....... 
L = [a,b,c,e,f,g,h]                  %   tedy průchod do šířky  . 

Cvičení 

 a)   Modifikujte dané procedury pro  procházení stromem tak , aby místo  vydávání  obsahu  jednotlivých  
uzlů  stromu  do  seznamu, volali v každém nějakou proceduru akce. 

 b)   Sestavte procedury na procházení grafu do hloubky a do šířky.  

 Vhodnou  reprezentaci  grafu   prologovským  termem  si  zvolte sami. Pozor v grafu mohou být cykly ! 

10.2. Binární vyhledávací stromy 

Nejčastěji  se  setkáváme  se  speciální  případem binárních stromu,  s  tzv.  binárními  vyhledávacími  stromy. 
Binární strom nazveme vyhledávacím, pokud pro každý uzel U tohoto stromu platí, že  hodnoty v  jeho levém  
podstromě jsou  menší než  U a hodnoty v jeho pravém podstromě jsou větší než U. 

Sestavme několik procedur pro práci s  binárními vyhledávacími stromy: 

%  ============================================================== 
%  prvek(?X,+Strom)       test, zda X je obsaženo ve stromě Strom 
%  ============================================================== 
prvek(X,t(_,X,_)).  %  vrchol stromu je jeho prvkem 
prvek(X,t(L,V,R)):- 
               X<V, %  X je menší než vrchol V, 
               prvek(X,L).  %  může tedy být jen v levém podstromě 
prvek(X,t(L,V,R)):- 
               X>V, %  X je větší než vrchol V, 
               prvek(X,R). %  může tedy být jen v pravém podstromě 

Cvičení 

Modifikujte  proceduru  tak,  aby  při  odmítání odpovědi na dotaz  prvek(-X,+Strom) vydala  všechny  
       prvky  stromu v rostoucím pořadí. 

%  ============================================================== 
%  vloz(+X,+Strom,-Strom1)     Strom1 vznikne vložením  prvku X do 
%                              stromu  Strom (pokud  tam již  byl, 
%                              bude Strom1 stejný jako Strom) 
%  ============================================================== 

vloz(X,nil,t(nil,X,nil)).  %  Vložení prvku X do prázdného stromu 
vloz(X,t(L,X,R),t(L,X,R)).  %  X je již vrcholem vstupujícího stromu 
vloz(X,t(L,V,R),t(L1,V,R)):- 
               X<V,  %  X je menší než vrchol stromu 
               vloz(X,L,L1).  %  L1 vznikne vložením prvku X  do levého podstromu L 
vloz(X,t(L,V,R),t(L,V,R1)):- 
               X>V,  %  X je větší než vrchol stromu 
               vloz(X,R,R1).  %  R1 vznikne vložením prvku X  do pravého podstromu R 

Cvičení 

  Lze proceduru  vloz použít ve  tvaru vloz(+X,-Strom,+Strom1) k vypouštění prvků z binárního 
vyhledávacího stromu?  Svoji odpověď odůvodněte! 

Cvičení 

  Sestavte procedury 
     vylejrost(+Strom,-Sezn)         vylejkles(+Strom,-Sezn)  , 
  které  vydávají   v  seznamu  Sezn  prvky   obsažené  v  binárním vyhledávacím stromě Strom 
     a) uspořádané od nejmenšího k největšímu       resp. 
     b) uspořádané od největšího k nejmenšímu 

Algoritmus vypuštění prvku  z  binárního vyhledávacího stromu  je poněkud obtížnější. 

�� Je-li vypouštěný  prvek  l is tem  stromu  (tj.  nemá-li  v něm ani jednoho  syna), je  jeho vypuštění  snadné -  
nikdo nemá  co  dědit. 

�� Má-li  vypouštěný  prvek  nejvýše jednoho  syna,   pak  tento  syn  zdědí postavení otce. 
�� Má-li vypouštěný prvek  oba syny, je  situace komplikovanější. Na místo vypouštěného  prvku  může  přijít  

buď nejmenší prvek z pravého podstromu nebo největší z levého. V našem programu  jsme si vybrali druhou 
možnost. Nejpravější prvek nemůže mít pravého syna - jde ho tedy vypustit jednoduše. 

Myslím,  že  mi  dáte  za  pravdu,  že  následující  program v Prologu  je  srozumitelnějším  (a  pochopitelně  i  
přesnějším) popisem tohoto algoritmu. 

%  ============================================================== 
%  vypust(+Prvek,+Vstrom,-Zbytek)   Zbytek je strom, který vznikne 
%                                   ze vstupního stromu Vstrom 
%                                   vypuštěním prvku Prvek 
%  ============================================================== 
vypust(X,nil,nil). %  vypuštění z prázdného stromu 
vypust(X,t(nil,X,R),R). %  vypuštění prvku, který nemá levého syna 
vypust(X,t(L,X,nil),L). %  vypuštění prvku, který nemá  pravého syna 

vypust(X,t(L,X,R),t(L1,Y,R)):-  %  vypouštěný prvek X, který má oba syny 
 nejprav(L,L1,Y).  %  bude nahrazen prvkem Y, který je nejpravějším   
    %  uzlem v levém podstromu L 
vypust(X,t(L,V,R),t(L1,V,R):-  
               X<V,  %  prvek X menší než je kořen stromu V 
               vypust(X,L,L1). %  se vypouští z levého podstromu 
vypust(X,t(L,V,R),t(L,V,R1):-  
               X>V,   %  prvek X větší než je kořen stromu V 
               vypust(X,R,R1). %  se vypouští z pravého podstromu R 

%  ============================================================== 
%             pomocná procedura pro vypust 
%  nejprav(+Vstrom,-Zbytek,-Nejpr)  Zbytek je strom, který vznikne 
%                                   ze  vstupního   stromu  Vstrom 
%                                   vypuštěním jeho  nejpravějšího 
%                                   prvku Nejpr 
%  ============================================================== 
nejprav(t(L,X,nil),L,X).  %  je-li pravý podstrom prázdný, je 
    %  nejpravějším prvkem kořen stromu 
nejprav(t(L,X,R),t(L,X,R1),Y):-  %  jinak je nejpravějším prvkem 
               nejprav(R,R1,Y).  %  nejpravější prvek pravého podstromu 

Binární vyhledávací stromy  jsou výhodnou datovou strukturou pro vyhledávání,  pokud jsou jejich podstromy  na 
všech úrovních, co nejvyváženější.  V nejhorším případě však  jde jen o složitěji realizovaný seznam (je-li u každého 
uzlu jeden podstrom prázdný). 

D o k o n a l e   v y b a l a n c o v a n é   binární   vyhledávací   stromy  jsou takové binární  vyhledávací 
stromy, ve  kterých se v  každém uzlu počet prvků v jeho pravém a levém podstromě liší nejvýše o jeden.  

Následující procedura s t a v í  z e  z a č á t k u  ( d a n é  d é l k y )  r o s t o u c í  p o s l o u p n o s t i  p r v ků  d o k o n a l e  
v y v á ž e n ý  s t r o m : 

%  ============================================================== 
%    postav(+N,+RostSez,-VybStrom,-Zbytek) 
%          Předpokládá, že RostSez je rostoucí seznam. 
%            Je-li délky alespoň N, postaví použití procedury 
%            z jeho prvních N prvků dokonale vybalancovaný 
%            strom. Je-li seznam kratší, postaví se strom ze 
%            všech prvků seznamu. Zbytek je nespotřebovaný 
%            zbytek seznamu 
%  ============================================================== 
postav(_,[],nil,[]).  %   vstupní seznam byl vyčerpán 
postav(0,T,nil,T).  %   do stromu již není třeba žádný prvek přidávat 
postav(N,In,t(L,V,R),Z):- 
          N1 is N-1,   %  v obou podstromech bude N1 prvků 
          NR is N1 div 2,  %  vpravo případně o 1 méně 
          NL is N1-NR,  %  než vlevo 
          postav(NL,In,L,[V|In1]), %  do levého přijde NL prvků 
          postav(NR,In1,R,Z). %  do pravého přijde NR prvků 

Cvičení 

Co se stane, odmítneme-li výsledný strom středníkem ?  Odstraňte tuto závadu procedury. 

11. Operátor řezu 

Všechny  součásti  programovacího  jazyka,  které jsme dosud probrali, patří (možná s výjimkou aritmetiky a 
jejího is) do tzv. č i s t é h o  P r o l o g u .   N e s n a ž í m e  s e  v  n ěm   n i j a k  o v l i vňo v a t  a l g o r i t m u s  
v y h o d n o c o v á n í   d o t a z ů .   Za  to   máme  jistotu  o  poměrně  těsné souvislosti  deklarativního  a  procedurálního  
významu programu.      Bohužel  však -  jak tomu  už v  programování bývá  - musíme často při řešení konkrétních úloh  
dělat kompromisy a z původních idejí slevovat. Nevystačíme proto jen s "čistým" Prologem. 

Ke zvýšení  efektivity prologovských programů  a k vyjádření některých  predikátů,  které  jsme  dosud  neuměli 
reprezentovat, slouží  standardní  prologovský  p r e d i k á t   ř e z u .   Značí se znakem vykř ičník  !   .  

P r e d i k á t    ř e z u    o k a m ž i t ě    u s p ě j e ,    a l e    p ř i   t o m   z a k á ž e  
 b a c k t r a c k o v á n í  p ř e s  s e b e  z p ě t .  

Např. mějme následující proceduru: 
          a(X,Y):- b(X,Z),!,c(Z,Y). 
          a(X,Y):- d(X,Y). 

Splní-li se cíl b(X,Z), splní se okamžitě i následující řez. Když  nyní dojde  k selhání  cíle c(Z,Y)  výpočet celé  
procedury skončí neúspěchem. Kdyby  v proceduře  řez nebyl,  zkoušela by  se alternativní unifikace cíle b(X,Z) a po  
jejím případném selhání by se zkoušel cíl d(X,Y) z druhé klauzule procedury. 

V  odstavci  6.1  jsme  sestrojili  proceduru prvek(X,Sezn), která při odpovídání na dotaz 
                       ?- prvek(-X,+Sezn) 

v případě odmítání výsledku středníkem (nebo při návratu přes ní, je-li  součástí těla  nějaké klauzule)  vydávala 
postupně všechny prvky seznamu  (pokud se v  seznamu opakovaly, pak  vícekrát). To nám může někdy vadit. 
Sestrojme její obměnu, která vydá pouze první prvek,  s použitím operátoru řezu: 

%  ============================================================== 
%     prvek1(-X,+L)   vydá první výskyt X v L 
%                     není-li X v L selže 
%  ============================================================== 
prvek1(X,[X|L]):- !.     %  je-li prvek nalezen, je zakázán návrat 
prvek1(X,[_|L]):-        %  nebyl-li nalezen X v hlavě (jinak se sem výpočet nedostane), 
          prvek1(X,L).   %  budeme ho hledat v těle 

Sestrojme proceduru,  která v l o ž í  d o   s e z n a m u  p r v e k  j e n  v  t o m  p ř í p a d ě ,  ž e  t a m  j i ž  n e b y l :  

%  ============================================================== 
%     pridej(+X,+L,-NL)   seznam NL vznikne ze seznamu L 
%                   přidání prvku X na jeho začátek ovšem jen 
%                   v tom případě, že X v L již není 
%  ============================================================== 
pridej(X,L,L):- 
  prvek(X,L), %  je-li X již prvkem L, nepřidám ho 
  ! . %  a zakáži návrat 
pridej(X,L,[X|L]).  %  X není prvkem L (jinak bych se 
   %  sem nedostal), mohu ho tedy přidat 

Cvičení 

    1) Jak se predikát pridej bude chovat při jiných dotazech, např. 
               pridej(-X,+L,-NL)  , atp. 

     2) Zkuste predikát pridej naprogramovat bez pomoci řezu ! 

11.1. Negace 

Již  v druhé  kapitole  jsme  narazili na  potřebu b i n á r n í h o  p r e d i k á t u   r u z n e ,  který  by  u s p ě l ,   k d y ž   
j e h o  a r g u m e n t y  n e j d o u  u n i f i k o v a t .  Pomocí řezu ho již sestavíme. 

Pokusme se vyjádřit v Prologu tvrzení 

" J a n a  m á  r á d a  m u ž e ,  a l e  n e  p l e š a t é "   .  

Bez  operátoru  řezu  to  nejde.  S  ním  a  se  standardním predikátem fail, který, je-li volán, okamžitě selže, ji sestavíme 
poměrně snadno: 

marada(jana,X):- 
  plesaty(X), %  je-li X plešaté uspěje, 
  !, %  zakáže návrat 
  fail. %  a selže. 
Marada(jana,X):-  %  k této klauzuli se výpočet dostane, pokud  X není plešaté,  
  muz(X).  %  je-li to muz, má ho Jana ráda 

Obdobně můžeme d e f i n o v a t  p o ž a d o v a n ý  p r e d i k á t  r u z n e . 

%  ============================================================== 
%       ruzne(X,Y)     uspěje, nejdou-li X a Y unifikovat 
%  ============================================================== 
ruzne(X,Y):- 
          X=Y,           %  jdou-li X a Y unifikovat 
          !, fail.       %  zakáži návrat a selžu 
ruzne(X,Y).              %  Sem se dostanu jen, nejdou-li X a Y 
                         %  unifikovat 

Ve většině implementací Prologu je takový predikát standardní a jmenuje se \= . 

Obecně můžeme definovat predikát n e g a c e  not  následovně: 

%  ============================================================== 
%     not(P)      uspěje, pokud se nepodaří cíl P splnit 
%  ============================================================== 
not(P) :- P, ! , fail. 
not(P). 

Označení predikátu negace není v jednotlivých implementacích Prologu  stejné. V  některých se  zapisuje jako  
predikát not(P), v jiných jako unární operátor bez závorek  - not P (tak ji budeme psát v dalším i my). 

V  některých  implementacích  Prologu  není  přípustné  psát argument některého  predikátu jako cíl, musí  být 
proto předchozí definice negace zapsána takto: 
     not(P) :- call(P), ! , fail. 
     not(P).                                        , 

kde  standardní  predikát  call  zprostředkuje  "volání argumentu P jako cíle". 

Je lepší používat pro vyjádření  negace operátor not, než ji opisovat pomocí  operátoru řezu. Lépe se  tak v našich 
programech vyznáme. 

V ý j i m k o u   z  tohoto  pravidla   jsou  však  k o n s t r u k c e   typu " p o d m i ňo v a c í h o  p ř í k a z u " ,  např.: 

a(X,Y):- not b(X),       %  pokud b(X) neplatí 
         c(X,Y).         %  pak c(X,Y) 
a(X,Y):- b(X),           %  pokud b(X) platí 
         d(X,Y).         %  pak d(X,Y) 

V nich se,  v případě, že b(X) platí,  vyhodnocuje tento cíl zbytečně  dvakrát,  jak  je  vidět  z  rozepsání  definice 
našeho predikátu a: 

a(X,Y):- neb(x),c(X,Y). 
a(X,Y):- b(X),d(X,Y). 
neb(X):- b(X),!,fail. 
neb(X). 

Proto je  efektivnější (i přehlednější)  tento tvar definice predikátu a(X,Y): 

a(X,Y):- b(X),!,         %  platí-li b(X), 
         d(X,Y).         %  pak d(X,Y) 
a(X,Y):- c(X,Y).         %  jinak, tj. neplatí-li b(X), c(X,Y). 

Cvičení 

Jaký je vlastně význam negace,  jak je definována v Prologu? Odpovídá  negaci,   jak  ji  známe   z  běžného 
        života,   logiky a matematiky ? 

Pokud jste  odpověděli, že ne, z  jiných než psychologických důvodů - Proč by se jinak ptal? , je to úspěch. 

verze 3.03   strana 43  



 

Rozdíl je následující: 

  V běžné  řeči, a především v klasické (jsou i jiné) logice , jsou 

          výrok A       i jeho negace         not A 

výroky  o témže  světě.  Výroky " je   hezká"  a  "není   hezká"  j sou hodnocením  ně jakého objektu.  

Naproti tomu v Prologu 

jehezka(X) je konstatováním vlastnosti objektu X, 

zatímco jeho negace 

not jehezka(X)   znamená jen, že z programu nelze odvodit platnost 
 predikátu jehezka(X). 

Je tedy negace výrok o programu,  ne o objektu X,  ten  (ta) X  může být  hezká -  ba překrásná  , ale 
někdo to zapomněl  zanést do  programu -  a už  platí not  hezka(X), a pro program  X hezká není. 

S  negací ve  smyslu Prologu  se setkáváme  v demokratických státech u soudu: 

Komu  jsme nedokázali,  že vrah  je, ten  vrahem (ve  smyslu práva) není. 

11.2. Důsledky použití řezu 

Jak  jsme  viděli,  umožňuje  nám  operátor  řezu 
�� o v l i vňo v a t  e f e k t i v i t u  p r o l o g o v s k ý c h  p r o g r a mů  

     (tím, že se nehledá zbytečně v některých větvích prostoru  úlohy), 
�� d e f i n o v a t  v z á j e m n ě  s e  v y l u č u j í c í  p o u ž i t í  j e d n o t l i v ý c h  k l a u z u l í  p r o c e d u r y , 
�� d e f i n o v a t  n e g a c i , 
�� . . .    . 

 Přes tyto  přínosy se doporučuje  u ž í v a t  h o  " s   r o z u m e m "  t j .  m á l o . Většinou totiž ničí deklarativní 
význam programu. 

Uvažujme dvě procedury 

     p :- a, b.                     q :- a, !, b. 
     p :- c.                        q :- c. 

Deklarativní význam procedury p je p <==> ( a & b) v c    , 
  ( kde v je logická spojka "vel", "or" - nebo ) . 

Deklarativní význam procedury q je  q <==> ( a & b ) v ( not a & c ) . 

Zaměníme-li  v obou  procedurách pořadí  klauzulí, dostáváme procedury: 

     p1 :- c.                        q1 :- c. 
     p1 :- a, b.                     q1 :- a, !, b. 

Deklarativní význam procedury p1 je stejný jako procedury p.  

Deklarativní význam procedury q1 je však jiný                q1 <==>   c v ( a & b ) . 

Někdy se rozlišují dva typy řezu: 
�� z e l e n ý  ř e z  -  je takové použití řezu,  jehož odstranění nemá vliv na  deklarativní  význam 

 programu  (odřezává  jen některé větve výpočtu) 
�� č e r v e n ý  ř e z  - je takové použití řezu, které deklarativní význam programu mění. 

Řez v procedurách  q a q1 je červený. 

  Č e r v e n ý m  ř e z ů m  b y c h o m  s e  r a d ě j i  m ě l i  v y h ý b a t  (ne vždy to však jde). 

12. TŘÍDĚNÍ SEZNAMŮ 

V  předchozím textu  jsme si  ukázali implementaci třídícího algoritmu   quicksort  v   Prologu.  Protože   jde  o  
rekursivní algoritmus, není obtížné jej "přepsat  do Prologu". Pokud jste si již zkusili  naprogramovat i jiné  třídící 
algoritmy, jistě  jste zjistili, že je to o něco obtížnější. 

V tomto  odstavci si ukážeme  implementace několika dalších třídících algoritmů a vrátíme se  i ke quicksortu. 
Povšimněte si, jakou roli v našich programech hraje operátor řezu. 

N a i v n í  t ř í d ěn í  

Tento algoritmus  je t a k  p r o s t o d u c h ý   (a tedy i  pomalý), že jej  uvádíme jen  jako o d s t r a š u j í c í   
p ř í p a d . Vychází  z myšlenky, systematického g e n e r o v á n í   v š e c h  p e r m u t a c í  vstupního  seznamu tak 
dlouho, dokud nenajdeme tu, která je uspořádaná. 

/* (velmi) naivní třídění 
   naive_sort(+L,-S)      postupně generuje backtrackováním 
                            všechny permutace seznamu L 
                            tak dlouho, dokud nevygeneruje 
                            uspořádanou 
   perm(+L,-P)            P je permutace seznamu L 
   ins(+X,+L,-S)          S vznikne vložením prvku X do seznamu L 
   sorted(+L)             testuje, zda seznam L je rostoucí 
*/ 

naive_sort(L,S):- perm(L,S),sorted(S). 

perm([],[]). 
perm([X|T],S):- perm(T,S1),ins(X,S1,S). 

ins(X,T,[X|T]). 
ins(X,[Y|T],[Y|T1]):- ins(X,T,T1). 

sorted([X,Y|Z]):- X<Y, sorted([Y|Z]). 
sorted([X]). 
sorted([]). 

T ř í d ěn í  v k l á d á n í m  

Tento  algoritmus třídí  rekursivně tak,  že do  setř íděného tě la  vstupního  seznamu  vloží   na  
správné  místo  hlavu tohoto seznamu.  

/* třídění vkládáním 
   insert(+X,+L,-S)  seznam S vznikne vložením prvku X na 
    správné místo do uspořádaného seznamu L 
   insert_sort(+[X|T],-S) nejprve se setřídí tělo T vstupního 
   seznamu, pak se do něj vloží na správné 
   místo hlava X vstupního seznamu  
*/ 

insert_sort([],[]). 
insert_sort([X|Tail],Sorted) :- 
          insert_sort(Tail,SortedTail), 
          insert(X,SortedTail,Sorted). 

insert(X,[Y|Sorted],[Y|Sorted1]) :- 
          X>Y , %  je-li X>Y musí být vložen do těla, 
          !,  %  jiná možnost není,proto řez 
          insert(X,Sorted,Sorted1). 
insert(X,Sorted,[X|Sorted]). 

B u b l i n k o v é  t ř í d ěn í  

Pravděpodobně   nejpopulárnějším   třídícím   algoritmem  je bublinkové  třídění, které  je založeno  na 
odstraňování  inverzí , tj. dvojic  sousedních prvků, které nejsou  správně uspořádány. 

/*  bublinkové třídění 
    swap(+L,-S)     uspěje pokud je v seznamu L inverze, 
                      seznam S potom vznikne odstraněním 
                      první z těchto inverzí 

    bubble_sort(+L,-S)  pokud je v seznamu L inverze, 
                          je odstraněna, je zakázáno vracení 
                          a na nový seznam je opět zavolán 
                          tentýž predikát 
                        pokud inverze v L není, pak tento seznam 
                          je již setříděný 
*/ 

bubble_sort(List,Sorted) :- 
 swap(List,List1),   %  ve vstupním seznamu byla odstraněna jedna inverze 
 !,    %  při navracení by byla obnovena, proto řez 
 bubble_sort(List1,Sorted). 
bubble_sort(Sorted,Sorted).  %  v seznamu není žádná inverze, je tedy setříděn 

swap([X,Y|Rest],[Y,X|Rest]) :- X>Y. %  odstranění inverze na začátku seznamu 
swap([Z|Rest],[Z|Rest1]) :- 
 swap(Rest,Rest1).  %  hledání a případné 
odstranění inverze v těle seznamu 

T ř í d ěn í  v ý b ě r e m  

Tento algoritmus  je založen na  tom, že se  vybere nejmenší prvek a  umístí se na první  místo setříděného 
seznamu. Po  té se totéž opakuje pro zbytek vstupního seznamu. 

/*  třídění výběrem 
   min(+L,-Min) Min je minimálním prvkem seznamu L 
   min(+L,+X,-Min) Min je minimum z čísla X a prvků seznamu L 
   select_sort(+L,-S) Hlavou seznamu se stane minimum seznamu L, 
        jeho tělem setříděný zbytek L (po odebrání 
     tohoto minima) 
*/ 

select_sort([],[]). 
select_sort(T,[X|S]):- min(T,X), 
                       del(X,T,T1), 
                       select_sort(T1,S). 

min([Z|T],X):- min(T,Z,X). 
min([Y|T],Z,X):- Y>=Z, min(T,Z,X). 
min([Y|T],Z,X):- Y<Z, min(T,Y,X). 
min([],X,X). 

del(X,[X|T],T):- !.   %  vypuštěním X výpočet okamžitě končí 
del(X,[Y|T],[Y|T1]):- del(X,T,T1). 

13. Reprezentace množin pomocí seznamů 

Poměrně    často   potřebujeme    v   programech    p r a c o v a t  s  ( k o n e č n ý m i )   m n o ž i n a m i .  Množiny   
můžeme  reprezentovat  mnoha různými   způsoby.  Při   porovnávání  jejich   vhodnosti  záleží především  na tom,  jaké 
prvky  množiny mohou  mít a jaké operace s množinami   chceme   provádět.  

  Nejjednoduším   způsobem,  jak reprezentovat v  Prologu množinu, je pracovat  s p r o s t ý m  s e z n a m e m  
j e j í c h  p r v ků . Přívlastek prostý znamená,  že se žádný prvek nesmí v seznamu vyskytovat vícekrát. 

Sestavíme  jako  programátorské  cvičení  některé jednoduché predikáty pracující s touto reprezentací množin. 

13.1. Množiny reprezentované jako prosté seznamy prvků 
/*  Množiny reprezentované jako prosté seznamy prvků 
 uprava(+Sezn,-Mnoz)      uspěje, je-li Sezn seznam, Mnoz je 
                            pak ekvivalentní prostý seznam tj. 
                            množina 
 vyhod(+H,+Sezn,-Sezn1)   uspěje, je-li Sezn seznam, Sezn1 z něj 
                            vznikne vypuštěním všech výskytů 
                            prvku H 
 vyhod1(+H,+Mnoz,-Mnoz1)  uspěje, je-li Mnoz seznam. 
                            Je-li H prvkem seznamu Mnoz, vznikne 
                              seznam Mnoz1 ze seznamu Mnoz 
                              vynecháním prvního výskytu prvku H, 
                            není-li H prvkem Mnoz, je Mnoz1 
                              totožné s Mnoz. 
                            Je-li tedy Mnoz množina, vznikne 
                            množina Mnoz1 případným vypuštěním 
                            (jediného) výskytu H 
 p_member(+H,+Mnoz,-Mnoz1)  uspěje, je-li H prvkem seznamu Mnoz. 
                              Seznam Mnoz1 vznikne ze seznamu 
                                Mnoz vypuštěním prvního výskytu 
                                prvku H; 
                              je-li tedy Mnoz množina, pak Mnoz1 
                                je tato množina bez prvku H 
 je_mnoz(+Sezn)           uspěje, je-li Seznam množinou (prostý) 
 rovny(+Mnoz1,+Mnoz2)     uspěje, pokud jsou Mnoz1 a Mnoz2 
                             rovny jako množiny 
 sjednoc(+M1,+M2,-Sjedn)  Sjedn je sjednocením množin M1 a M2 
                            počítá se jako  zřetězení a následná 
                            úprava 
 uprava1(+Sezn,-Mnoz)     uspěje, je-li Sezn seznam. 
                            V novém seznamu Mnoz má každý prvek 
                              o jeden výskyt méně než v původním 
                              seznamu Sezn. 
                            Pokud je v seznamu  Sezn každý prvek 
                              nejvýše dvakrát, je pak Mnoz 
                              ekvivalentní prostý seznam 
 sjednoc1(+M1,+M2,-Sjedn)   Sjedn je sjednocením množin M1 a M2 
                               počítá rekursí přes množinu M1 
 prunik(+M1,+M2,-Prunik)    Průnik množin M1 a M2 
 rozdil(+M1,+M2,-Rozdil)    Rozdíl množin M1 a M2 
*/ 

  uprava([H|T],[H|T1]):- vyhod(H,T,T2),uprava(T2,T1). 
  uprava([],[]). 

  vyhod(H,[H|T],T1):- !, vyhod(H,T,T1). 
  vyhod(H,[K|T],[K|T1]):- vyhod(H,T,T1). 
  vyhod(H,[],[]). 

  vyhod1(H,[H|T],T):- ! . 
  vyhod1(H,[K|T],[K|T1]):- vyhod1(H,T,T1). 
  vyhod1(H,[],[]). 

  p_member(H,[H|T],T):- !. 
  p_member(H,[K|T],[K|T1]):- p_member(H,T,T1). 

  je_mnoz([X|T]):- not p_member(X,T,_), je_mnoz(T). 
  je_mnoz([]). 

  rovny([X|T],M):- p_member(X,M,M1),rovny(T,M1). 
  rovny([],[]). 

  sjednoc(M1,M2,S):- conc(M1,M2,S1), uprava1(S1,S). 

  uprava1([H|T],[H|T1]):- vyhod1(H,T,T2),uprava1(T2,T1). 
  uprava1([],[]). 

  sjednoc1([H|T],M,[H|T1]):- vyhod1(H,M,M1), sjednoc1(T,M1,T1). 
  sjednoc1([],M,M). 

  prunik([H|T],M,[H|T1]):- p_member(H,M,M1), !, prunik(T,M1,T1). 
  prunik([H|T],M,T1):- prunik(T,M,T1). 
  prunik([],_,[]). 

  rozdil([H|T],M,T1):- p_member(H,M,M1),!,rozdil(T,M1,T1). 
  rozdil([H|T],M,[H|T1]):- rozdil(T,M,T1). 
  rozdil([],_,[]). 

  conc([H|T],S,[H|U]):- conc(T,S,U). 
  conc([],S,S). 

Cvičení 

Vytvořte  predikát podmn(Mnoz1,Mnoz2),který uspěje právě když je Mnoz1 podmnožinou množiny  
          Mnoz2. 

13.2. Množiny reprezentované jako rostoucí seznamy prvků 
Asi vás  napadlo, že by  se práce s  množinami zjednodušila, kdybychom  o jejich  reprezentacích předpokládali  

víc, například to,  že  p r v k y   v   s e z n a m e c h   j s o u   u s p o ř á d á n y   o d   n e j m e n š í h o   k  n e j v ě t š í m u .   
Největší výhodou  bude skutečnost,  že p a k   m á  k a ž d á  m n o ž i n a  p r á v ě  j e d n u  r e p r e z e n t a c i . 

/*  Množiny reprezentované jako rostoucí seznamy 
 je_mnozr(+Sezn)             uspěje, je-li Sezn množinou, 
                               tj. je rostoucí 
 upravar(+Sezn,-Mnoz)        uspěje, je-li Sezn seznam, Mnoz je 
                               pak ekvivalentní rostoucí seznam 
 rovnyr(+Mnoz1,Mnoz2)        uspěje, pokud jsou množiny Mnoz1 
                               a Mnoz2 rovny (jako množiny) 
 p_memberr(+H,+Mnoz,-Mnoz1)  uspěje, je-li H pvrkem množiny Mnoz, 
                               Mnoz1 je tato množina bez prvku H 
 sjednocr(+M1,+M2,-Sjedn)    Sjedn je sjednocením množin M1 a M2 
 prunikr(+M1,+M2,-Prunik)    Průnik množin M1 a M2 
 rozdilr(+M1,+M2,-Rozdil)    Rozdíl množin M1-M2 */ 

  je_mnozr([X,Y|T]):- X<Y, je_mnozr([Y|T]). 
  je_mnozr([X]). 
  je_mnozr([]). 

  upravar(Sezn,Mnoz):- uprava(Sezn,PrSezn), %  z minulého příkladu 
                       sort(PrSezn,Mnoz).   %  libovolné třídění 

  rovnyr(M1,M2):- M1=M2. 

  prunikr([X|T],[X|S],[X|U]):- prunikr(T,S,U). 
  prunikr([X|T],[Y|S],U):- X<Y, prunikr(T,[Y|S],U). 
  prunikr([X|T],[Y|S],U):- X>Y, prunikr([X|T],S,U). 
  prunikr([],_,[]). 
  prunikr(_,[],[]). 

  sjednocr([X|T],[X|S],[X|U]):- sjednocr(T,S,U). 
  sjednocr([X|T],[Y|S],[X|U]):- X<Y, sjednocr(T,[Y|S],U). 
  sjednocr([X|T],[Y|S],[Y|U]):- X>Y, sjednocr([X|T],S,U). 
  sjednocr([],S,S). 
  sjednocr(S,[],S). 

  p_memberr(H,[H|T],T):- !. 
  p_memberr(H,[K|T],[K|T1]):- K<H, p_memberr(H,T,T1). 

  rozdilr([X|T],[X|S],U):- rozdilr(T,S,U). 
  rozdilr([X|T],[Y|S],[X|U]):- X<Y, rozdilr(T,[Y|S],U). 
  rozdilr([X|T],[Y|S],U):- X>Y, rozdilr([X|T],S,U). 
  rozdilr([],_,[]). 
  rozdilr(T,[],T). 

14. Vstup a výstup  

Vstupy a výstupy jsou  v programování velmi důležité. Přesto můžeme  pozorovat tendenci,  že při  výkladu 
programování  se jim věnuje  čím  dál  tím  menší  pozornost  (srovnej  např.  FORTRAN a Pascal). Pro výklad  Prologu 
i experimentování s  ním v jistém smyslu vstup  a výstup  vůbec  nepotřebujeme,  protože máme  s programem 
interaktivní kontakt bez programování vstupů a výstupů. 

V Prologu  je však možno - jako v každém plnoprávném programovacím jazyku - naprogramovat i klasický  vstup 
ze souboru a výstup do souboru. 

 Při   startu   programu    je   a k t u á l n í    v s t u p    nastaven z klávesnice,  a k t u á l n í  v ý s t u p  na obrazovku. 
Tyto vstupy a výstupy označuje atom user. 

Ke změně aktuálního vstupu a výstupu slouží predikáty: 

see(F)  nastaví aktuální vstup ze souboru F 
seen(F)  uzavře vstupní soubor F a obnoví aktuální vstup z klávesnice  
  ( jako by zavolal see(user)  ) 
seeing(-F) dotaz na aktuální vstupní soubor 
tell(F) nastaví aktuální výstup do souboru F 
told(F) uzavře výstupní soubor F a obnoví aktuální výstup  na obrazovku 
   ( jako by zavolal tell(user)  ) 
telling(-F)   dotaz na aktuální výstupní soubor 

Základním typem  vstupu (resp. výstupu)  v Prologu je  vstup (resp. výstup) termů. Slouží k němu predikáty: 

read(?T)  přečte  z   aktuálního  vstupu   jeden  term  (ukončený tečkou) 
  a unifikuje jej s argumentem T 
write(+T) vystoupí instance termu T s právě platnými hodnotami substitucí za 
  proměnné v termu T obsažené 

verze 3.03   strana 49  



 

Dalším typem  je vstup a  výstup znaků. Slouží  k němu např. tyto predikáty: 

get0(?Z)  přečte jeden znak z aktuálního vstupu, 
   uspěje pokud jej lze unifikovat s termem Z 
get(Z) chová se stejně, pouze ignoruje (přeskakuje), řídící znaky (ASCII < 32) 
put(Z) výstup znaku do aktuálního výstupu (ne řídící znaky) 
tab(N) vystoupí N mezer (N přirozené číslo) 
nl  přechod na novou řádku ( pascalské writeln )   . 

Dříve než uvedeme několik příkladů, dovolte malé intermezzo. 

14.1. Programování cyklů 

Tento odstavec  vlastně do kapitoly  o vstupech a  výstupech nepatří, zařadili jsme ho sem proto,  že se nám cykly 
budou při programování vstupů a výstupů hodit. 

I  když je  Prologu vlastní  rekursivní programování,  někdy potřebujeme  naprogramovat  i  cyklus.  
Nejjednodušším způsobem je použití standardního nulárního predikátu repeat. 

%  ============================================================== 
%     repeat         nulární predikát okamžitě uspěje 
%                      a uspěje vždy i při návratu 
%  ============================================================== 
repeat. 
repeat:- repeat. 

a )  c y k l u s  r e p e a t - u n t i l  

obdobu tohoto druhu cyklu můžeme naprogramovat takto: 
          repeat, C1, C2, C3, .... ,Cn ,Podm , ! . 

Splňování posloupnosti cílů  C1, C2, C3, .... ,Cn  se opakuje tak dlouho, dokud není splněn cíl Podm. 

%  ============================================================== 
%    vstup     přečte ze vstupu jeden term 
%              pokud to není přirozené číslo menší než 100, 
%              opakuje výzvu a čtení 
%  ============================================================== 
vstup:- repeat, 
        write('Zadej přizené číslo menší než 100:'),  %  výzva 
        read(N),  %  přečtení termu ze vstupu 
        integer(N), %  uspěje je-li term N celé číslo (standardní predikát) 
        N>0,   %  uspěje, je-li číslo, které vstoupilo 
        N<100,    %  větší než 0 a menší než 100 
        ! . 

b )  n e k o n e čn ý  c y k l u s  r e p e a t - f a i l  

Splňování posloupnosti cílů 
          repeat, C1, C2, C3, .... ,Cn , fail . 

vede k nekonečnému opakování splňování posloupnosti cílů 
                      C1, C2, C3, .... ,Cn   . 

Odůvodněte proč. 

c )  f o r  c y k l u s  

Cyklus  se   vzrůstající  hodnotou  parametru   jde  poměrně přirozeně  naprogramovat  pomocí   rekurse.  
Příkladem  může  být procedura seznam(N,S) z kapitoly 8. 

Přímá konstrukce for cyklu je  poněkud násilná, asi je lepší používat rekursi. Poslouží nám však jako pěkné cvičení. 

Cvičení 

Uvažujme predikát for definovaný takto: 
 for(I,I,J):- I>J,!,fail. 
 for(I,I,J). 
 for(I,J,K):- J1 is J+1, for(I,J1,K). 

Dokažte, že procedura 
 p(I,Dolni,Horni,Telo):- for(I,Dolni,Horni), call(Telo), fail. 

 p(_,_,_,_). 

způsobí  opakovaná  splňování  cíle   Telo  s  rostoucí  hodnotou proměnné I počínaje od hodnoty Dolni  
 až po hodnotu Horni. 

Cvičení Modifikujte predikát for, aby šlo zadat i krok. 

14.2. Příklady 

a )  p r e d i k á t  t a b  

Naprogramujme standardní predikát tab 

%  ============================================================== 
%    tab(+N)  předpokládá, že parametr N je přirozené číslo 
%               na standardní výstup vystoupí N mezer 
%  ============================================================== 
tab(0). 
tab(N):- put(' '), N1 is N-1, tab(N1). 

b )  v i z u a l i z a c e  b i n á r n í h o  s t r o m u  n a  t i s k á r n ě  

%  ============================================================== 
%      ukaz(+Strom)    vytiskne binární strom Strom, 
%               "položený na bok" : kořen vlevo, listy vpravo 
%      uk(+Strom,+Odsaz,+D)  vytiskne strom Strom s kořenem 
%               odsazeným o Odsaz mezer od levého kraje, 
%               podstormy budou odsazeny o D mezer dál 
%  ============================================================== 
ukaz(Strom):- uk(Strom,0,2). 
uk(nil,K,_):- 
     tab(K),write(nil),nl. 
uk(t(L,V,R),K,D):- 
          K2 is D+K,               %   odsazení podstromů 
          uk(R,K2,D),              %   pravý podstorm 
          tab(K),write(V),nl,      %   kořen 
          uk(L,K2,D).              %   levý podstrom 

c )  k o p í r o v á n í  s o u b o r u  F 1  d o  s o u b o r u  F 2  

%  ============================================================== 
%     kop(+F1,+F2)  zkopíruje soubor F1 do souboru F2 
%                      obnoví původní nastavení vstupních souborů 
%  ============================================================== 
kop(F1,F2):- 
     seeing(In), %  zjištění a uschování aktuálních 
     telling(Out), %  I/O souborů 
     see(F1), %  otevření zdrojového souboru 
     tell(F2), %  otevření cílového souboru 
     repeat,  %  opakuj 
     read(Term), %  přečti Term 
     ((  %  druhá závorka není nutná 
       Term=end_of_file, %  test konce souboru 
       !,  %  ukončení, je-li konec souboru 
       told,seen, %  uzavření souborů (je často zbytečné, 
   %  provede se automaticky při otevření nových) 
       see(In),tell(Out) %  obnovení původních I/O souborů 
      )  %  končí akce na konci souboru 
      ;  %  alternativa - soubor F1 nekončí 
      (  %  závorka opět není nutná 
       write(Term), %  výpis Termu do F2 
       fail  %  navracení k počátku cyklu 
      ) 
     ). 

d )  P ř e č t e n í  ř á d k y  z  k l á v e s n i c e  a  j e j í  v ý p i s  n a  o b r a z o v k u  

%  ============================================================== 
%     ctiradku(-S)  přečte z klávesnice do seznamu S jednu řádku 
%                     ukončenou znakem CR (ASCII=13) 
%     pisradku(+S)  vypíše na obrazovku seznam znaků S 
%  ============================================================== 
ctiradku(Sez):- 
     get0(X),       %  přečtení znaku X 
     (X=13,         %  je-li X CR 
      Sez=[],       %  ukonči tvorbu seznamu 
      !             %  zařízni 
      ;             %  alternativa - x není CR 
      ctiradku(S),  %  přečti tělo seznamu 
      Sez=[X|S]     %  X bude hlavou výsledného seznamu 
     ). 
pisradku([]):- !,nl.    %  výstup prázdného seznamu - 
                        %    odřádkování a konec 
pisradku([X|Z]):-       %  výstup neprázdného seznamu - 
     put(X),            %  výstup prvního znaku 
     pisradku(Z).       %  výstup zbylých znaků 

14.3. Definování operátorů 
V syntaxi Prologu jsme se kromě obvyklého predikátového zápisu setkali na několika místech i s použitím 

operátorů, například aritmetických, s operátorem unifikace =, operátorovým zápisem negace a pod. Operátory jsou i 
symboly :- , čárka či středník. 

 Operátorový zápis je v Prologu vlastně j e n  z á l e ž i t o s t í  v s t u p u  a  v ý s t u p u . Proto nečiní problém dát 
programátorovi možnost definovat si vlastní operátory. To může značně pomoci čitelnosti programů, protože jsme z 
řady oblastí na užívání operátorů zvyklí. Ukažme si, jak se to dělá. 

K definování operátorů slouží standardní ternální predikát  op: 

op (?Priorita, ?Asociativita, ?Jmeno)            . 

Začneme posledním t ř e t í m  argumentem Jmeno , jehož hodnotou je  a t o m  o z n a ču j í c í  p ř í s l u š n ý  
o p e r á t o r .   

P r v n í  argument Priorita je „číselný“ a u d á v á  p r i o r i t u  d a n é h o  o p e r á t o r u . Číselná vyjádření priorit 
se mohou u různých implementací lišit, vždy však platí, že čím větší číslo, tím operátor váže méně - má nižší prioritu. 
Abychom tento argument mohli rozumně využít, musíme znát  číselná vyjádření priorit, která naše implementace užívá.  

Hodnotou druhého argumentu Asociativita může být jeden z následujících, na první pohled tajemných atomů: 
xfx       xfy      yfx pro binární infixové operátory 
fx        fy pro unární prefixové operátory 
xf        yf pro unární postfixové operátory.  

Tímto způsobem udáváme počet argumentů (jeden nebo dva) a u unárních argumentů i pozici operátoru. Symbol f 
v těchto atomech reprezentuje operátor, symboly x a y jeho operandy. Mezi x a y je následující rozdíl: na místě 
symbolu y může stát jakýkoli operand, na místě x pouze operand, který není vytvořen pomocí operátoru se stejnou nebo 
vyšší prioritou (připomínám vyšší priorita - nižší číslo) než má právě definovaný operátor. Asi Vám to stále není jasné, 
ukažme si tedy několik příkladů. 

Aritmetické operátory by mohly být definovány například takto: 

op(500, yfx, +). op(500, yfx, -).    % binární operátory typu sčítání, 
op(450, fx, +). op(450, fx, -).  % unární plus a minus, 
op(400, yfx, *). op(400, yfx, /).  % binární operátory typu násobení. 

Z těchto definic je jasné, že 

 term  znamená protože 

a + b + c ( a + b ) + c první argument smí mít stejnou prioritu, druhý musí mít větší 
a + b * c a + ( b * c ) operátor * má větší prioritu než operátor +  
+ a - - b ( + a ) - (  - b ) unární znaménka jsme definovali s větší prioritou než binární - 
+ + a je nepřípustné znaménko má asociativitu fx, kdyby byla fy, bylo by přípustné 

Příklad nám jen posloužil jako ilustrace. Většinou mají znaménka stejnou prioritu jako binární plus a minus. 

Pomocí klauzulí  

op(506,fx,p).   op(505,fy,q).   op(503,yf,r).   op(504,xf,s). 

jsme definovali čtyři unární operátory a, b, c, d. První dva, p a q, jsou prefixové, druhé dva postfixové. Můžeme tedy 
psát termy 

p arg , q arg  , arg r , arg s  , 

 v nichž arg je atom, který je argumentem. Můžeme psát i následující termy: 

q q arg   % operátor q na výsledek operátoru q na argument arg, 
p p arg % není přípustné, p má asociativitu fx 
p q arg % operátor p na výsledek operátoru q na argument arg 
p arg r r % operátor p na výsledek dvojnásobného použití operátoru r na argument arg 

    Standardní operátory jsou definovány naprosto stejným způsobem. Potřebujete-li zjistit, jaké operátory jsou 
právě definovány, můžete se to dovědět dotazem 

?- op (Priorita, Asociativita, Jmeno).             
musíme pochopitelně získané odpovědi odmítat středníkem, abychom dostali postupně všechny definované operátory 
(elegantněji by to šlo pomocí operátoru bagof, se kterým se seznámíme v další kapitole). 

Kdybychom k programu našich klepen přidali klauzuli  

op( vhodné_číslo, xfx,  rodic ). , 

mohli bychom místo predikátového zápisu rodic(X,Y) používat operátorový  X rodic Y. 

15. Natažení programu do paměti, program jako 
"databáze" 

Zatím  jsme si  neřekli, jak  program v  Prologu "načteme do paměti", abychom s ním mohli pracovat. 

Slouží nám k tomu dva standardní predikáty: 

consult(F)  načte program ze souboru F do paměti, 
   pokud je již v paměti program uložen, 
  přidá přečtené klauzule na konec programu 
reconsult(F) načte program ze souboru F do paměti  
    pokud je již v paměti program uložen, vymaže z něj ty procedury, 
  které jsou  obsaženy v souboru a nahradí je jejich verzemi ze souboru 

Je  tedy   třeba  p o u ž í v a t   r e c o n s u l t  v ž d y   p o   o p r a v ě   s o u b o r u . Načteme-li  opravený  soubor  
pomocí  consult,  přidají se konzultované klauzule za stávající - získáme tedy v paměti zcela nesmyslný  program. 
A nejen to - tento program se bude většinou chovat stejně jako program před opravou, neboť se nové verze  změněných  
procedur neuplatní,  protože   budou  v  programu  až  za původními. Tato chyba se začátečníkovi špatně odhaluje - 
v souboru je všechno v pořádku. 

Některé  překladače  nemají  oba predikáty  consult a reconsult. Obvykle pak predikát consult pracuje jako reconsult. 

Program v Prologu obsahuje typicky  také d a t a  - jako v našem programu SK z  kapitoly 2, j e  d o b r ý m  
z v y k e m   u c h o v á v a t  j e  v  j i n é m  s o u b o r u  n e ž  " v l a s t n í  p r o g r a m "  a  b ě h e m  v ý p o č t u  j e  
„ n a k o n z u l t o v a t “  pomocí predikátu consult (resp.reconsult). Usnadní to práci s jinými daty. 

Velmi často lze místo consult(F) napsat jen [F], a dokonce můžeme napsat stručné [ F1,F2,F3] ve významu 
posupného provádění tří cílů consult(F1), consult(F2), consult(F3). 

Soubor, který konzultujeme nemusí obsahovat jen klauzule tvarů, které jsme se dosud naučili, ale také tzv. 
„příkazy“. Formálně jsou to klauzule s „prázdnou hlavou“. Mají tvar 
   :- tělo . , 

kde tělo je posloupnost klauzulí. Tyto příkazy se při konzultování hned provádějí. Velmi často to jsou příkazy consult 
nebo jiné inicializační akce.  

Skládá-li se například náš program z několika částí umístěných v různých souborech, dejme tomu 
ZDROJ1.PL , ZDROJ2.PL  a   ZDROJ3.PL, můžeme do prvního z nich přidat klauzuli 

:- consult(‘ZDROJ2.PL’), consult(‘ZDROJ3.PL’). 

Tím se při konzultování programu z prvního souboru „natáhnou“ do paměti i zbylé dva. Chceme-li některé jiné, máme 
snadnou práci s opravou.  

Protože  na program  V Prologu   se jde  dívat jako  na data obecného  interpretu  prologovských  programů,  říká  
se mu někdy " d a t a b á z e " .   A Prolog  nám poskytuje  i prostředky,  jak do  této databáze - programu p ř i d á v a t  
n o v é  k l a u z u l e ,  č i  j i n é  v y ř a z o v a t .  Při tom přidávané klauzule mohou  být vytvořeny až při výpočtu 
programu. Mohou tedy hrát i roli jakýchsi "globálních proměnných", ve kterých si můžeme pamatovat mezivýsledky. 

I když se v obecnosti bez těchto prostředků obejít nelze, je d o b r é   s n a ž i t   s e   j i m   v y h ý b a t .  Nejen,  že 
kazí význam programu (modifikace programu během jeho výpočtu  se dnes již nenosí ani v asembleru)  a  ztěžují  jeho  
ladění,  ale  v  Prologu m a j í  n a v í c  n e g a t i v n í   v l i v  n a   r y c h l o s t  v ý p o č t u   - paradoxně  tím více,  čím 
rychlejší kompilátor máme. 

V našem textu  mnoho příkladů na užití  těchto prostředků úmyslně  neuvádíme, protože  umožňují programovat  
v Prologu "jako v Pascalu" s použitím "globálních  proměnných" a obcházet tak to, co jsme vás chtěli naučit. 

Prostředky pro aktualizaci databáze jsou: 

assert(+K) přidá klauzuli K na konec programu v paměti 

assertz(+K) ekvivalentní s assert(K) 

verze 3.03   strana 54  



 

asserta(+K)  přidá klauzuli K na začátek programu v paměti 

retract(?K) odstraní z programu v paměti první výskyt klauzule, která  
  jde unifikovat s argumentem K 

retractall(?H)  odstraní z programu v paměti všechny klauzule, jejichž  
  hlava unifikuje s termem H 

16. Několik standardních predikátů a  operátorů 
V  této  kapitole  uvedeme  bez  velkých  komentářů  několik další  standardních  predikátů  a  operátorů,  které  

můžete  při programování potřebovat. Na další, ale i na přesný význam těchto, se  musíte   podívat  do  manuálu   
implementace,  kterou  budete používat. 

Hned po informaci, jak nějaký program spustit, je nejdůležitější i n f o r m a c e ,  j a k  z  n ě j  o d e j í t . Na prolog 
většinou neplatí žádné exit, quit či end, ale německé halt. 

halt ukončení práce s Prologem 

Většina překladačů má celou unárních standardních predikátů, které t e s t u j í ,  z d a  a r g u m e n t  m á  d a n o u  
v l a s t n o s t . 

atom(X)  uspěje, je-li X atom 
atomic(X)  uspěje, je-li X atom nebo číslo 
integer(X)  uspěje, je-li X integer 

var(X) uspěje, je-li X dosud nekonkretizovaná proměnná 
nonvar(X) uspěje, není-li X proměnná nebo je-li to proměnná, která byla již konkretizována 

Další dva predikáty umožňují  t e s t o v a t ,  z d a  d v a  t e r m y  j s o u  e k v i v a l e n t n í   (před unifikací, ne po ní) 

X==Y uspěje, jsou-li X a Y ekvivalentní termy 
X\==Y uspěje, nejsou-li X a Y ekvivalentní termy 
  (při těchto predikátech nedochází k unifikaci) 

např. 
?- X==Y. 

no               %  X a Y jsou dvě různé proměnné 
?- X=Y, X==Y. 

X=_12  Y=_12   %  unifikací při splňování cíle X=Y byly  proměnné X a Y ztotožněny 

Další predikát umožňuje p ř í s t u p  k  a r g u m e n tům  s t r u k t u r y  a  n a o p a k  k o n s t r u k c i  s t r u k t u r y  z   
a r g u m e n tů : 

Term =.. L   uspěje, má-li seznam L jako hlavu hlavní funktor F termu Term a  
    jako tělo seznam argumentů funktoru F v termu T 
   (může se používat obousměrně) 

Příklad 

Sestrojme predikát , který provede s u b s t i t u c i  j e d n o h o  t e r m u  d r u h ý m  z a  v š e c h n y  v ý s k y t y  v  
d a n é m  t e r m u .  

/* subst( Vstup, Co, Cim, Vystup) 
provede v termu Vstup substituci za všechny výskyty (pod)termu Co termem Cim ; výsledkem je term Vystup 

*/ 

subst(Vstup,Vstup,Vystup,Vystup). % Substituuje se za celý term 
subst(Vstup, _ , _ ,Vstup):- 
  atomic(Vstup).  % Jinak, není-li Vstup struktura, substituovat nelze   

subst(Vstup,Co,Cim,Vystup):-  
  Vstup =.. [Funktor|Argumenty] , % rozebereme strukturu 
  subst_v_seznamu(Argumenty,Co,Cim,Vyst_Argumenty), 
   % provedeme substituce na všech argumentech 
  Vystup =..[Funktor|Vyst_Argumenty]. % a zase ji složíme zpět 

subst_v_seznamu([Hlava|Ocas],Co,Cim,[NovaHlava|NovyOcas]) :- 
  subst(Hlava,Co,Cim,NovaHlava), 
   subst_v_seznamu(Ocas,Co,Cim,NovyOcas). 
subst_v_seznamu([ ],_ ,_ ,[ ]). 

O b d o b n o u  r o l i  hrají i další dva standardní predikáty 

functor(?Term,?F,?Arita)    uspěje, je-li F hlavní funktor termu Term o aritě Arita 
arg(+N,?Term,?Arg)             uspěje, je-li Arg N-tým argumentem termu Term 

Další predikát umožňuje  v y t v á ř e t  z  a t o mů  z n a k o v é  ř e t ě z c e  a  n a o p a k :  

name(?Atom,?List)  uspěje, je-li List seznam ASCII kódů znaků atomu Atom 

Uvědomme si, že znakové řetězce jsou v Prologu ekvivalentní se seznamy ASCII kódů jednotlivých znaků. 

Příklad 

Předpokládejme, implementace, se kterou pracujeme dává k dispozici má k dispozici standardní predikát dos, který 
zavolá příkaz operačního systému (vpodstatě každý překladač nějaký takový predikát má). Sestavíme predikát e(F), 
který zavolá váš oblíbený editor (dejme tomu, že se volá příkazem dosu „turbo <jmeno souboru>“) na soubor F a po 
provedení oprav nakonzultuje novou verzi programu. 

e(F):- name(F,RetF),  % F atom, RetF znakový řetězec 
  conc(RetF,“.PL“,RetFsPrip), % RetFsPrip je úplný název souboru s příponou PL 
  conc(“turbo “,RetFsPrip,RetPrikaz), % konstrukce řetězce příkazu dosu 
  name(Prikaz,RetPrikaz), % převod řetězce na atom 
  dos(Prikaz),  % volání editoru příkazem operačního systému 
   % zde pracuje editor, po jeho ukončení se bude splňovat další predikát  
  reconsult(F).  % rekonzultování opraveného souboru 

Jistě jste si uvědomili, že musíte na konci práce s editorem opravovaný soubor uložit. 

Nakonec jsme si nechali dva velmi užitečné predikáty, které umožňují jednoduše „ n a s b í r a t “  v š e c h n y  t e r m y  
m a j í c í  n ě j a k o u  v l a s t n o s t .  Tato vlastnost může být velmi složitá. Začneme prvním z nich, druhý 
predikát je jen jeho modifikací. 

bagof(+Term,+Cil,-Seznam)  Seznam je seznam všech instancí termu Term, pro něž je splněn cíl Cil 

S naším programem SK (Spolku Klepen - podle nového pravopisu už  "evropsky"  Klepen  s  velkým  K)  bychom 
dostali následující odpovědi: 

?- bagof(X,rodic(karel,X),L). 
    X=_0038  L=[zuzana,alfred,emil] 

?- bagof(X,manzdite(X),L). 
    X=_0038  L=[josef,hugo,zuzana,alfred,eva,katka,emil] 

?- bagof(dvoj(X,Y),vmanzelstvi(X,Y),L). 
     X=_0038  L=[dvoj(adam,eva),dvoj(karel,helena), 
                 dvoj(zibrid,kunhuta),dvoj(jan,lucie), 
                 dvoj(jose,katka),dvoj(eva,adam), 
                 dvoj(helena,karel),dvoj(kunhuta,zibrid), 
                 dvoj(lucie,jan),dvoj(katka,jose)] 

Nyní již lze uvést i druhý predikát 

 setof(+Term,+Cil,-Seznam)  Totéž jako bagof, jen v seznamu Seznam je každý prvek právě jednou 
  (jde o množinu reprezentovanou prostým seznamem) 

Ještě na jednu skutečnost musíme upozornit. Oba predikáty se totiž chovají v případě, že neexistuje instance 
prvního argumentu,  pro kterou je cíl v druhém argumentu splněn. V tomto případě není „výsledkem“ ve třetím 
argumentu prázdný seznam, jak byste možná očekávali, ale oba predikáty v tomto případě neuspějí.. 

17. Závěrem - aneb co všechno jsme nestihli 

     Proti  původnímu  záměru  se  rozsah  textu neúměrně zvětšil a stal se z  něho začátek ucelené učebnice 
Prologu.  Přesto v něm mnoho chybí. Namátkou jmenuji 

          - práce s datovými strukturami 
          - ukázky typických úloh, které se pomocí Prologu řeší 
          - více o tom, jak se Prolog překládá 
          - více o logickém programování 
          - alespoň jedna větší úloha 
          - ..... 

Proč  je pro  vás dobré  se s  Prologem seznámit,  i když ho třeba nebudete nikdy používat nebo učit ? 

  - Pomůže vám pochopit, co je to rekurse. 
  - Dá vám nový pohled na programování. Co  je podstatné a co jen technika. 
  - Snad i víc 

